

Ovo je zahvala.

Sadržaj

1. Uvod	2
2. Coq	3
2.1. Što je Coq?	3
2.2. Programiranje u Coqu	5
2.3. Hijerarhija tipova	8
2.4. Propozicije i tipovi, dokazi i termi	10
2.5. Ograničenja tipskog sustava	11
3. Logika prvog reda s induktivnim definicijama	14
3.1. Sintaksa	15
3.2. Semantika	22
3.3. Standardni modeli	25
3.4. Sistem sekvenata s induktivnim definicijama	29
3.5. Adekvatnost	34
4. Ciklički dokazi	35
5. Zaključak	36
Literatura	37
Sažetak	39
Abstract	40

1. Uvod

2. Coq

U ovom poglavlju dajemo pregled visoke razine programskog sustava Coqa. Prvo ćemo objasniti što je uopće Coq, u kojem je kontekstu nastao, i od kojih komponenti se sastoji. Zatim ćemo dati kratak pregled programiranja u Coqu, nakon čega ćemo se baviti naprednijim konceptima i spomenuti neka ograničenja. Za širi opseg gradiva, čitatelja upućujemo na knjige *Coq'Art* [1], *Software Foundations* [2, 3, 4] i *Certified Programming with Dependent Types* [5] te na službenu dokumentaciju [6].

2.1. Što je Coq?

Alat za dokazivanje Coq¹, punog naziva *The Coq Proof Assistant*, programski je sustav pomoću kojeg korisnici mogu dokazivati matematičke tvrdnje, a može se koristiti i kao funkcijski programski jezik sa zavisnim tipovima. Alat se temelji na λ -računu i teoriji tipova, a prva je inačica implementirana 1984. godine [6] Ovaj rad koristi inačicu 8.18 iz rujna 2023. godine.

Program Coq može se pokrenuti u interaktivnom ili u skupnom načinu rada. Interaktivni način rada pokreće se naredbom `coqtop`, a korisniku omogućuje rad u ljusci sličnoj ljuskama `bash` i `python`. Interaktivna ljuska (također poznata pod imenom *toplevel*) služi unosu definicija te iskazivanju i dokazivanju tvrdnji. Skupni način rada pokreće se naredbom `coqc`, a korisniku omogućuje semantičku provjeru i prevođenje izvornih datoteka u strojno čitljive formate. Kod formaliziranja i dokazivanja, korisnik će najčešće koristiti interaktivni način rada, po mogućnosti kroz neku od dostupnih razvojnih okolina.²

¹<https://coq.inria.fr/>

²Autor rada koristio je paket *Proof General* za uređivač teksta *Emacs*. Druge često korištene okoline su *VsCoq* i *CoqIDE*.

Kao programski jezik, Coq se sastoji od više podjezika različitih namjena, od kojih spominjemo *Vernacular*, *Gallinu* i *Ltac*.

Vernacular („govorni jezik”) je jezik naredbi kojima korisnik komunicira sa sustavom (i u interaktivnom i u skupnom načinu rada); svaka Coq skripta (datoteka s nastavkom `.v`) je niz naredbi. Neke od najčešće korištenih naredbi su `Check`, `Definition`, `Inductive`, `Fixpoint` i `Lemma`. Pomoću naredbi za iskazivanje tvrdnji, kao što je `Lemma`, Coq ulazi u način dokazivanja (*proof mode*).

Gallina je Coqov strogo statički tipiziran specifikacijski jezik. Dokazi svih tvrdnji predstavljeni su interno kao programi u Gallini. Kako se glavnina programiranja u Coqu svodi upravo na programiranje u Gallini, posvećujemo joj idući odjeljak.

Ltac je Coqov netipizirani jezik za definiciju i korištenje taktika. Taktike su pomoćne naredbe kojima se u načinu dokazivanja konstruira dokaz. Može se reći da je Ltac jezik za metaprogramiranje Galline. Primjeri taktika su `intros`, `destruct`, `apply` i `rewrite`.

Pogledajmo ilustrativan primjer.

```
1  Lemma example_lemma: 1 + 1 = 2.  
2  Proof.  
3    cbn. reflexivity.  
4  Qed.
```

Ključne riječi `Lemma`, `Proof` i `Qed` dio su Vernaculara, izraz `example_lemma: 1+1=2` dio je Galline, a pomoćne naredbe `cbn` i `reflexivity` dio su Ltaca.

Jezgra je programskog sustava Coq algoritam za provjeru tipova (*type checking*) — svaka tvrdnja koja se dokazuje iskazana je tipovima. Ostatak sustava u načelu služi za knjigovodstvo i poboljšanje korisničkog iskustva.³ Nužno je da jezgra sustava bude relativno mala kako bismo se mogli uvjeriti u njenu točnost. U suprotnom, možemo li biti sigurni da su naše dokazane tvrdnje doista istinite?

Prve inačice Coqa implementirale su samo račun konstrukcija [7] — proširenje λ -računa polimorfnim i zavisnim tipovima te tipskim konstruktorima. Kasnije je dodana podrška za induktivno i koinduktivno definirane tipove [8, 9], a danas se može reći da Coq implementira polimorfni kumulativni račun induktivnih konstrukcija [10]. Coq se, osim kao dokazivač teorema, može koristiti i za programiranje sa zavisnim tipovima. U

³I jezgra i ostatak sustava implementirani su u OCamlu.

toj sferi konkuriraju jezici Agda⁴, Idris⁵ i Lean⁶. Coq se između njih ističe po usmjerenosti prema dokazivanju, posebno po korištenju taktika (jezik Ltac) i nepredikativnoj sorti Prop (o kojoj će kasnije biti riječi). Još jedna prednost Coqa je mehanizam *ekstrakcije* pomoću kojeg korisnik može proizvoljnu funkciju prevesti u jezik niže razine apstrakcije.⁷ Mehanizam ekstrakcije nije dokazano točan, no poželjno je da izvorne funkcije budu ekvivalentne ekstrahiranim pa se radi na verifikaciji ekstrakcije [10].

2.2. Programiranje u Coqu

Gallina je funkcijski programski jezik, što znači da su funkcije prvoklasni objekti — one mogu biti argumenti i povratne vrijednosti drugih funkcija. Dodatno, iteracija se ostvaruje rekurzijom te ne postoje tradicionalne varijable, već se koriste nepromjenjiva (*immutable*) imena. Za uvod u funkcijsko programiranje, čitatelja upućujemo na knjigu *Programming in Haskell* [11]. Primjeri koje ćemo vidjeti u ostatku ovog odjeljka oslanjanju se na tipove i funkcije definirane u Coqovoj standardnoj knjižnici.⁸

Gallina je strogo statički tipiziran jezik, što znači da se svakom termu prilikom prevođenja dodjeljuje tip⁹. Naredbom `Check` možemo provjeriti tip nekog terma ili doznati da se termu ne može dodijeliti tip. Dalje u radu pod „term” mislimo na dobro formirane terme, odnosno na one kojima se može dodijeliti tip. Kažemo da je term *stanovnik* tipa koji mu je dodijeljen. Za tip kažemo da je *nastanjen*, odnosno *nenastanjen*, ako postoji, odnosno ne postoji, stanovnik tog tipa. Kako su u Coqu i tipovi termi, radi razumljivosti i zvučnosti umjesto „tip tipa” kažemo „sorta tipa”.

Kao i u ostalim jezicima, kod programiranja u Coqu korisnik se oslanja na dostupne primitivne izraze, od kojih su najvažniji:

- `forall` za konstrukciju funkcijskih tipova i zavisnih produkata,
- `match` za rad sa stanovnicima induktivnih tipova te
- `fun`, `fix` i `cofix` za definiciju funkcija.

Naredbom `Inductive` definira se *induktivni* tip te se automatski za njega generiraju

⁴<https://wiki.portal.chalmers.se/agda/>

⁵<https://www.idris-lang.org/>

⁶<https://lean-lang.org/>

⁷Trenutno su podržani Haskell, OCaml i Scheme.

⁸<https://coq.inria.fr/library/>

⁹Tipovi su kolekcije objekata na kojima je moguće provoditi srodne operacije.

principi *indukcije* i *rekurzije*.

```
1 Inductive nat : Set :=  
2 | 0 : nat  
3 | S : nat -> nat.
```

Ovim kodom definirali smo tri terma:

- `nat` (tip prirodnih brojeva) je term sorte `Set`,
- `0` (broj nula) je term tipa `nat` i
- `S` (funkcija sljedbenika) je term tipa `nat → nat`.

Za term `nat` kažemo da je konstruktor tipa (*type constructor*), a za terme `0` i `S` kažemo da su konstruktori objekata (*object constructors*).

Jedna od osnovnih naredbi za imenovanje novih terma je naredba `Definition`.

```
1 Definition negb (b : bool) : bool :=  
2 match b with  
3 | false => true  
4 | true  => false  
5 end.
```

U gornjem kodu definirana je funkcija `negb` čiji se argument `b` tipa `bool` destrukture te se vraća njegova negacija, također tipa `bool`. Važno je napomenuti da izrazi koji počinju s `match t`, gdje je `t` stanovnik tipa `T`, moraju imati po jednu granu za svaki konstruktor tipa `T`.¹⁰ U ovom su primjeru konstante `false` i `true` jedini konstruktori tipa `bool`. Funkcija `negb` je tipa `bool → bool`.

```
1 Definition mult_zero_r : Prop := forall (n : nat), n * 0 = 0.
```

Ovdje je definirana propozicija (tip) imena `mult_zero_r` kao tvrdnja univerzalno kvantificirana po prirodnim brojevima.

Rekurzija nad induktivnim tipovima može se ostvariti naredbom `Fixpoint`, koja u pozadini koristi naredbu `Definition` te izraz `fix`.

¹⁰U tandemu s uvjetom strukturalne rekurzije, ovime je osigurana totalnost svake funkcije.

```

1 Fixpoint plus (n m : nat) {struct n} : nat :=
2   (* Definition plus := fix plus (n m : nat) {struct n} := *)
3   match n with
4   | 0 => m
5   | S n' => S (plus n' m)
6   end.

```

U ovom primjeru definirana je funkcija `plus` koja prima dva argumenta tipa `nat`. Funkcija je rekurzivna s obzirom na prvi argument što je vidljivo iz oznake `{struct n}`. Napominjemo da su induktivni tipovi dobro utemeljeni, to jest svaki term induktivnog tipa je konačan.

Osim induktivnih, u Coqu postoje i koinduktivni tipovi, koji nisu dobro utemeljeni, zbog čega za njih nije moguće definirati principe indukcije i rekurzije. Umjesto rekurzije, koinduktivni tipovi koriste se u korekurzivnim funkcijama. Standardan primjer koinduktivnog tipa je beskonačna lista.

```

1 Set Primitive Projections.
2 CoInductive Stream (A : Type) := Cons {
3     hd : A;
4     tl : Stream A;
5     }.

```

Ovime smo definirali familiju tipova `Stream` indeksiranu tipskom varijablom `A`. Svaki `Stream` ima glavu i rep koji je također `Stream`.

Stanovnici koinduktivnih tipova konstruiraju se korekurzivnim funkcijama naredbom `CoFixpoint`, koja u pozadini koristi naredbu `Definition` te izraz `cofix`.

```

1 CoFixpoint from (n : nat) : Stream nat := Cons _ n (from (S n)).
2   (* Definition from := cofix from (n : nat) := Cons _ n (from (S n)) *)

```

Ovime je definirana funkcija `from` koja za ulazni argument `n` vraća niz prirodnih brojeva od `n` na dalje.

Razlika induktivnih i koinduktivnih tipova može se sumirati epigramom:

„Induktivni tipovi su domene rekurzivnih funkcija, koinduktivni tipovi su kodomene korekurzivnih funkcija”.

Time se želi reći da se termi induktivnih tipova destruktuiraju u rekurzivnim funkcijama, dok se termi koinduktivnih tipova konstruira u korekurzivnim funkcijama.

2.3. Hijerarhija tipova

U usporedbi s tradicionalnim programskim jezicima, Coqov tipski sustav je ekspresivniji jer dopušta tipove koji mogu ovisiti o termima. Takvi tipovi se u Coqu konstruira izrazom oblika `forall`. Primjer jednog takvog tipa je „lista duljine n ”, gdje je n neki prirodan broj. Njegovi su stanovnici n -torke, a on ovisi o stanovniku drugog tipa (u ovom slučaju, o n tipa `nat`).


Spomenuli smo da su i tipovi termi te im se može dodijeliti sorta. Postoji li najveća sorta, odnosno postoji li tip `Type` čiji su stanovnici svi tipovi? Prisjetimo se, svaki term ima svoj tip. Kada bi takav `Type` postojao, tada bi vrijedilo `Type : Type`, što može dovesti do paradoksa samoreferenciranja.¹¹ Takav dokazivač teorema bio bi inkonzistentan te bismo njime mogli dokazati kontradikciju, čime dokazivač efektivno gubi svoju svrhu. Umjesto jedne „velike” sorte `Type`, u Coqu postoji rastući niz sorti `Typen` za sve prirodne brojeve n , takav da vrijedi `Typen : Typem` kad god vrijedi $n < m$. Ilustracija ove **kumulativne hijerarhije tipova** prikazana je na slici 2.1. Dvije najvažnije sorte u Coqu su `Set` i `Prop`.

Naziv `Set` sinonim je za sortu `Type0`, a njene stanovnike zovemo **malim tipovima**. Primjerice, tipovi `nat` i `bool` su mali tipovi. Dodatno, tipovi funkcija koje primaju i vraćaju male tipove su također mali tipovi. Također su i produkti, sume, liste i stabla malih tipova ponovo mali tipovi. Intuitivno se može reći da su mali oni tipovi s čijim se stanovnicima može efektivno računati. Stanovnike malih tipova nazivamo **programima**.

Stanovnici sorte `Prop` su **propozicije** (izjave). Za razliku od programa, s propozicijama ne možemo efektivno računati, ali ih možemo dokazivati. Stanovnici propozicija su njihovi **dokazi**.

Za dokazivače teorema, poželjna je mogućnost definicije predikata (propozicija) nad proizvoljnim tipovima. Zbog toga u Coqu prilikom definicije terma sorte `Prop` možemo

¹¹U naivnoj logici to je Epimenidov paradoks („Ova je rečenica lažna.”), u naivnoj teoriji skupova to je Russellov paradoks, u naivnoj teoriji tipova to je Girardov paradoks.



Slika 2.1. Kumulativna hijerarhija tipova

kvantificirati po proizvoljno velikim tipovima, što uključuje i sortu `Prop`. Tako je ovdje `not` predikat nad sortom `Prop`.

1 Definition not (P : Prop) := forall (Q : Prop), P -> Q.

Valja primijetiti da kod definicije predikata `not` kvantificiramo po `Prop`, no za svaku propoziciju P je term `not P` također tipa `Prop`, pa domena funkcije `not` uključuje sam term `not P`. Ovaj stil kvantifikacije omogućuje nam definiciju proizvoljnih propozicija i propozicijskih veznika. Kažemo da je sorta `Prop` **nepredikativna**. S druge strane, sorta `Set` je **predikativna**, to jest *ne dopušta* kvantifikaciju po `Set` i ostalim proizvoljno velikim tipovima. Kada bi sorta `Prop` bila predikativna, uopće ne bismo mogli definirati predikat `not`.

2.4. Propozicije i tipovi, dokazi i termi

U decimalnom zapisu broja π , barem jedna znamenka pojavljuje se beskonačno mnogo puta. Doista, kada bi se svaka znamenka pojavljivala samo konačno mnogo puta, broj π bio bi racionalan. Međutim, nije jasno *koja* znamenka ima to svojstvo. Možda ih ima više. Štoviše, vjerujemo da su *sve* znamenke takve. Da bismo odgovorili na to pitanje, morali bismo prebrojiti sve znamenke broja π , što nije moguće u konačno mnogo koraka.

Sličnim pitanjima bavili su se logičari dvadesetog stoljeća. *Klasični* logičari bi gornju tvrdnju smjesta prihvatili, dok bi *konstruktivisti* tražili konkretnu znamenku. Između ostalog, ovakva razmatranja rezultirala su fundamentalnim uvidom u povezanost programiranja i dokazivanja. Naime, želimo li dokazati konjunkciju, dovoljno je zasebno dokazati njene konjunkte. S druge strane, želimo li konstruirati par objekata, dovoljno je zapakirati prvi i drugi objekt u konstruktor para. Na sličan način, želimo li dokazati implikaciju, dovoljno je pretpostaviti njen antecedent te pomoću njega dokazati konzekvens. Ako pak želimo konstruirati funkciju, smijemo uzeti njen argument i pomoću njega konstruirati povratnu vrijednost. Dodatno, nemoguće je dokazati laž, a istina trivijalno vrijedi. S druge strane, ako induktivni tip nema konstruktore, onda je nenastanjen jer nije moguće definirati vrijednost tog tipa. Ako pak tip ima barem jedan konstruktor bez argumenata, tada postoje i njegovi stanovnici. Kroz ove primjere vidimo fenomen **Curry–Howardove korespondencije**, koju možemo sažeti epigramom:

„Propozicije su tipovi, dokazi su programi.”

Time se dokazivanje svodi na programiranje. Pogledi na dvije strane ovog novčića mogu se vidjeti u tablici 2.1.

Za bolju ilustraciju, prikazujemo princip matematičke indukcije u Coqu. Prisjetimo se, za proizvoljni predikat P na prirodnim brojevima, princip matematičke indukcije glasi:

$$P(0) \wedge \forall n, P(n) \rightarrow P(n + 1) \rightarrow \forall n, P(n).$$

Tvrdnju dokazujemo analizom broja n . Ako je $n = 0$, tvrdnja slijedi iz *baze* indukcije. Ako je pak $n = n' + 1$ za neki n' , tada rekurzivno konstruiramo dokaz za $P(n')$, a konačna tvrdnja slijedi primjenom *koraka* indukcije na rekurzivno konstruirani dokaz.

Dokazivanje	Programiranje
propozicija	tip
dokaz	program
laž	prazan tip
istina	nastanjen tip
konjunkcija	produktni tip
disjunkcija	zbrojni tip
implikacija	funkcijski tip
univerzalna kvantifikacija	zavisni produkt
egzistencijalna kvantifikacija	zavisna suma

Tablica 2.1. Sličnosti dokazivanja i programiranja

```

1 Definition nat_ind (P : nat -> Prop)
2   (baza : P 0)
3   (korak : forall n, P n -> P (S n))
4 : forall n, P n :=
5 fix F (n : nat) : P n :=
6   match n with
7   | 0 => baza
8   | S n' => korak n' (F n')
9 end.

```

Za term `nat_ind` kažemo da je dokazni term (*proof term*) za tvrdnju matematičke indukcije. Princip matematičke indukcije je samo poseban slučaj **principa indukcije**, koji Coq automatski generira za svaki induktivno definiran tip pri njegovoj definiciji.

2.5. Ograničenja tipskog sustava

Kao što smo već vidjeli, Coqov tipski sustav je izražajniiji od tipskih sustava uobičajenih programskih jezika. Međutim, kako bi se sačuvala poželjna svojstva algoritma provjere tipova, ipak se tipski sustav mora ograničiti.

Uvjet pozitivnosti odnosi se na definiciju induktivnih i koinduktivnih tipova. Ovo ograničenje zabranjuje *negativne* pojave tipa kojeg definiramo u argumentima njegovih konstruktora.

```

1 Inductive Lam :=
2 | LamVar (n : nat)
3 | LamApp (M N : Lam)
4 | LamAbs (M : Lam -> Lam).

```

Pokretanje primjera iznad rezultira greškom `Non strictly positive occurrence of "Lam" in "(Lam -> Lam) -> Lam"` — drugim riječima, tip `Lam` se javlja negativno u konstruktoru `LamAbs`, odnosno kao argument funkcije koja je parametar konstruktora. Ovaj uvjet štiti korisnika od inkonzistentnosti, a za točnu definiciju pozitivnosti čitate-lja upućujemo na dokumentaciju.¹² Uz uvjet pozitivnosti za induktivne tipove vezan je **uvjet strukturalne rekurzije**. Ovim uvjetom osigurava se totalnost rekurzivno defini-rane funkcije tako da se argument po kojem je funkcija rekurzivna strukturalno smanjuje u svakom koraku rekurzije (funkcija se poziva samo na pravom podtermu originalnog ar-gumenta).

Uvjet produktivnosti odnosi se na definiciju korekurzivnih funkcija, a dualan je uvjetu strukturalne rekurzije. Ovaj uvjet također štiti korisnika od inkonzistentnosti, a glasi: svaki korekurzivni poziv smije se pojaviti samo kao izravni argument konstruktora koinduktivnog tipa čiji element definiramo (poziv funkcije mora stvoriti pravi nadterm originalnog poziva).¹³ Zbog tog uvjeta, iduća definicija nije moguća.

```

1 Set Primitive Projections.
2 CoInductive NatStream := {
3     nat_hd : nat;
4     nat_tl : NatStream;
5 } .
6
7 CoFixpoint foo : NatStream := foo.
```

Greška koju sustav javlja glasi `Unguarded recursive call in "foo"`, što znači da se korekurzivni poziv `foo` ne javlja kao izravni argument konstruktora. S druge strane, definicija

```

1 CoFixpoint bar : NatStream := { | nat_hd := 0; nat_tl := bar | }.
```

je sasvim legalna.

Posljednje ograničenje koje spominjemo vezano je uz irelevantnost dokaza (*proof ir-relevance*). Naime, mnogi teoremi mogu se dokazati na više načina, ali pojedini dokaz (dakle, postupak kojim smo od pretpostavki došli do konkluzije) *nije bitan*. Matemati-

¹²<https://coq.inria.fr/doc/v8.18/refman/language/core/inductive.html#well-formed-inductive-definitions>

¹³<https://coq.inria.fr/doc/v8.18/refman/language/core/coinductive.html#co-recursive-functions-cofix>

čarima su bitni samo iskaz teorema i činjenica da se teorem *može dokazati*. Sam postupak dokazivanja smatra se „implementacijskim detaljem”. Upravo zato analiza dokaza ima smisla samo kada se dokazuje, ali ne i kada se programira. U našoj terminologiji to znači da se *pattern matching* nad dokazima smije provoditi samo kod definiranja terma sorte Prop. U suprotnom, mogli bismo definirati programe koji ovise o *konkretnom dokazu*, umjesto o iskazanom teoremu. **Ograničenje eliminacije propozicije** nastalo je radi omogućavanja ekstrakcije — svi termi sorte Prop se „brišu” prilikom prevođenja iz Coqovog tipskog sustava u tipske sustave niže razine apstrakcije. Kada ovog ograničenja ne bi bilo, ekstrakcija u jednostavnije jezike naprosto ne bi bila moguća, jer bi prilikom ekstrakcije bilo nužno zadržati i sve dokaze uz svu kompleksnost njihovih tipova.

3. Logika prvog reda s induktivnim definicijama

U ovom poglavlju predstavljamo glavne rezultate diplomskog rada: formalizaciju logike prvog reda s induktivnim definicijama FOL_{ID} te dokaznog sustava $LKID$, koje je prvi uveo Brotherston [12]. Definicije, leme i dokazi u ovom poglavlju preuzete su iz Brotherstonove disertacije [13]. Za općeniti uvod u logiku čitatelja upućujemo na knjigu *Matematička logika* [14].

Prvo ćemo definirati sintaksu i semantiku logike FOL_{ID} , nakon čega ćemo definirati njene standardne modele. Zatim ćemo prikazati dokazni sustav $LKID$ te konačno dokazati dio adekvatnosti sustava $LKID$ s obzirom na standardnu semantiku, što je ujedno i glavni rezultat ovog diplomskog rada. **Misao:** *Ovo ću preformulirati ako uspijem dokazati zadnje pravilo.*

Svaka definicija i lema u ovom poglavlju bit će popraćena svojom formalizacijom u Coqu. Jedan je od ciljeva diplomskog rada prikazati primjene Coqa u matematici, zbog čega leme nećemo dokazivati „na papiru”, već se dokaz svake leme može pronaći u repozitoriju rada.¹ Zainteresiranom čitatelju predlažemo interaktivni prolazak kroz dokaze lema.

Prije no što krenemo na formalizaciju, valja prokomentirati odnos matematičkog i Coqovog vokabulara što se tiče riječi „skup”. U matematici pojam „skup” može imati dva značenja; prvo se odnosi na skupove kao *domene diskursa*, dok se drugo odnosi na skupove kao *predikate*, odnosno podskupove domene diskursa. Primjerice, skup prirodnih brojeva \mathbb{N} je domena diskursa kada je riječ o svim prirodnim brojevima te zbog toga pišemo $n \in \mathbb{N}$ umjesto $\mathbb{N}(n)$. S druge strane, skup svih parnih prirodnih brojeva E je

¹TODO: repo link, na kraju

podskup skupa \mathbb{N} , a može se interpretirati kao predikat na prirodnim brojevima te možemo pisati $E(n)$ umjesto $n \in E$. U Coqu se skupovi kao domene diskursa formaliziraju tipovima sorte Set^2 , dok se skupovi kao predikati formaliziraju funkcijama iz domene diskursa u sortu Prop . Na primjer, tip prirodnih brojeva nat je sorte Set , a predikat Nat.Even je tipa $\text{nat} \rightarrow \text{Prop}$.

3.1. Sintaksa

Kao i u svakom izlaganju logike, na početku je potrebno definirati sintaksu.

Definicija 1. *Signatura prvog reda s induktivnim predikatima* (kratko: signatura), u oznaci Σ , je skup simbola od kojih razlikujemo *funkcijske*, *obične predikatne* i *induktivne predikatne* simbole. Mjesnost simbola reprezentiramo funkcijom iz odgovarajućeg skupa simbola u skup \mathbb{N} , a označujemo ju s $|f|$ za funkcijske, odnosno s $|P|$ za predikatne simbole.

```

1  Structure signature := {
2    FuncS : Set;
3    fun_ar : FuncS -> nat;
4    PredS : Set;
5    pred_ar : PredS -> nat;
6    IndPredS : Set;
7    indpred_ar : IndPredS -> nat
8  }.

```

Primjer 1. Funkcijski simboli Peanove aritmetike su redom simbol nule o (mjesnosti 0), simbol sljedbenika s (mjesnosti 1) te simboli $+$ za zbroj i \cdot za umnožak (mjesnosti 2).

```

1  Inductive Func__PA :=
2  | PA_zero
3  | PA_succ
4  | PA_add
5  | PA_mult.
6  Definition fun_ar__PA (s : Func__PA) : nat :=
7    match s with
8    | PA_zero => 0
9    | PA_succ => 1
10   | PA_add  => 2
11   | PA_mult => 2
12   end.

```

²Ili općenito kao tipovi sorte Type .

Simbol jednakosti (= (mjesnosti 2)) je jedini običan predikatni simbol Peanove aritmetike.

```

1 Inductive Pred__PA := PA_eq.
2 Definition pred_ar__PA (s : Pred__PA) : nat :=
3   match s with
4   | PA_eq => 2
5   end.

```

Definiramo induktivne predikatne simbole N , E i O koje ćemo interpretirati redom kao predikate „biti prirodan broj”, „biti paran broj” te „biti neparan broj”.

```

1 Inductive IndPred__PA :=
2   | PA_Nat
3   | PA_Even
4   | PA_Odd.
5 Definition indpred_ar__PA (s : IndPred__PA) : nat :=
6   match s with
7   | PA_Nat => 1
8   | PA_Even => 1
9   | PA_Odd => 1
10  end.

```

Konačno, *proširena Peanova signatura*, s oznakom Σ_{PA} , je signatura za Peanovu aritmetiku prvog reda s prethodno definiranim induktivnim predikatnim simbolima.

```

1 Definition  $\Sigma\_PA$  : signature
2   := { |
3     FuncS := Func__PA;
4     fun_ar := fun_ar__PA;
5     PredS := Pred__PA;
6     pred_ar := pred_ar__PA;
7     IndPredS := IndPred__PA;
8     indpred_ar := indpred_ar__PA;
9   | }.

```

U ostatku poglavlja promatramo jednu proizvoljnu, ali fiksiranu signaturu Σ . Fiksiranje nekog proizvoljnog objekta je česta pojava u matematici, prvenstveno zato što fiksirane argumente ne trebamo spominjati eksplicitno. Coq omogućuje fiksiranje naredbom `Context`, pod uvjetom da se korisnik nalazi u okolini `Section`.³ Većina definicija i lema u ovom radu su napisane upravo unutar takvih okolina.

³<https://coq.inria.fr/doc/v8.18/refman/language/core/sections.html>

Definicija 2. *Varijabla* je prirodan broj. *Skup svih terma* konstruiramo rekurzivno na način:

1. svaka varijabla je term;
2. ako je f funkcijski simbol mjesnosti n te su t_1, \dots, t_n termi⁴, onda je $f(t_1, \dots, t_n)$ također term.

```

1 Inductive term : Set :=
2   | var_term : var -> term
3   | TFunc : forall (f : FuncS  $\Sigma$ ), vec term (fun_ar f) -> term.

```

Uobičajene prezentacije logike prvog reda za skup varijabli uzimaju proizvoljan skup \mathcal{V} , no za formalizaciju je pogodniji skup prirodnih brojeva \mathbb{N} . Umjesto eksplicitne kvantifikacije po nekoj varijabli v , implicitno ćemo kvantificirati po varijabli 0. Ovaj pristup kvantifikaciji⁵, imena „de Bruijnovo indeksiranje”, bitno olakšava rad sa supstitucijama, a uveden je u članku [15]. O samoj implementaciji de Bruijnovog indeksiranja više se može pročitati u knjizi *Types and Programming Languages* [16]. Za potrebe ovog rada koristili smo program *Autosubst2*⁶ [17, 18] za automatsko generiranje tipova terma i formula te pripadajućih funkcija supstitucija i pomoćnih lema.

Princip indukcije za term potrebno je ručno definirati. Naime, induktivni tip term je *ugniježđen* po konstruktoru TFunc zato što se javlja oмотan oko drugog induktivnog tipa⁷ kao argument. Za ugniježdene induktivne tipove, Coq generira neprikladne principe indukcije jer ne zna kako izraziti tvrdnju „predikat vrijedi za sve ugniježdene elemente.”

```

1 Lemma term_ind
2   : forall P : term  $\Sigma$  -> Prop,
3     (forall v, P (var_term v)) ->
4     (forall f args, (forall st, V.In st args -> P st) ->
5                       P (TFunc f args)) ->
6     forall t : term  $\Sigma$ , P t.

```

⁴Primijetimo, broj terma ovisi o mjesnosti funkcijskog simbola. U Coqovoj implementaciji ovog „konstruktor” možemo vidjeti da je on zavisnog tipa.

⁵Ili općenitije, vezivanju varijabli.

⁶<https://github.com/uds-psl/autosubst2>

⁷Ovdje vec.

Definicija 3. Skup svih varijabli terma t , u oznaci $TV(t)$, konstruiramo rekurzivno na način:

1. $TV(v) := \{v\}$ za varijablu v i
2. $TV(f(t_1, \dots, t_n)) := \bigcup_{1 \leq i \leq n} TV(t_i)$ za n -mjesni funkcijski simbol f i terme t_1, \dots, t_n .

```

1 Inductive TV : term -> var -> Prop :=
2 | TVVar : forall v, TV (var_term v) v
3 | TVFunc : forall f args v st, V.In st args ->
4               TV st v -> TV (TFunc f args) v.

```

Definicija 4. Skup svih formula konstruiramo rekurzivno na način:

1. ako je Q (obični ili induktivni) predikatni simbol mjesnosti n te su t_1, \dots, t_n termi, onda je $Q(t_1, \dots, t_n)$ atomarna formula;
2. ako je φ formula, onda su $\neg\varphi$ i $\forall\varphi$ također formule;
3. ako su φ i ψ formule, onda je $\varphi \rightarrow \psi$ također formula.

```

1 Inductive formula : Set :=
2 | FPred (P : PredS  $\Sigma$ ) : vec (term  $\Sigma$ ) (pred_ar P) -> formula
3 | FIndPred (P : IndPredS  $\Sigma$ ) : vec (term  $\Sigma$ ) (indpred_ar P) -> formula
4 | FNeg : formula -> formula
5 | FImp : formula -> formula -> formula
6 | FAll : formula -> formula.

```

Ostale veznike definiramo kao sintaksne pokrate.

```

1 Definition FAnd ( $\varphi \psi$  : formula) : formula := FNeg (FImp  $\varphi$  (FNeg  $\psi$ )).
2 Definition FOr ( $\varphi \psi$  : formula) : formula := FImp (FNeg  $\varphi$ )  $\psi$ .
3 Definition FExist ( $\varphi$  : formula) : formula := FNeg (FAll (FNeg  $\varphi$ )).

```

Primjer 2. U proširenoj Peanovoj signaturi, svojstvo „svaki prirodan broj je paran ili neparan” možemo izraziti formulom $\forall x, N(x) \rightarrow E(x) \vee O(x)$.

```

1 Definition every_nat_is_even_or_odd : formula  $\Sigma$ _PA :=
2   FAll
3     (FImp
4       (@FIndPred  $\Sigma$ _PA PA_Nat ([var_term 0]))
5       (FOr
6         (@FIndPred  $\Sigma$ _PA PA_Even ([var_term 0]))
7         (@FIndPred  $\Sigma$ _PA PA_Odd ([var_term 0])))).

```

Ovdje vidimo učinak de Bruijnovog indeksiranja na formule; umjesto eksplicitnog navođenja varijable x kod kvantifikacije, implicitno kvantificiramo po varijabli 0.

Definicija 5. Skup slobodnih varijabli formule φ , u oznaci $FV(\varphi)$, konstruiramo rekurzivno na način:

1. $FV(P(u_1, \dots, u_n)) := \bigcup_{1 \leq i \leq n} TV(u_i)$,
2. $FV(\neg\varphi) := FV(\varphi)$,
3. $FV(\varphi \rightarrow \psi) := FV(\varphi) \cup FV(\psi)$,
4. $FV(\forall\varphi) := \{v \mid v + 1 \in FV(\varphi)\}$.

```

1 Inductive FV : formula -> var -> Prop :=
2   | FV_Pred : forall R args v st,
3     V.In st args -> TV st v -> FV (FPred R args) v
4   | FV_IndPred : forall R args v st,
5     V.In st args -> TV st v -> FV (FIndPred R args) v
6   | FV_Imp_l : forall F G v, FV F v -> FV (FImp F G) v
7   | FV_Imp_r : forall F G v, FV G v -> FV (FImp F G) v
8   | FV_Neg : forall F v, FV F v -> FV (FNeg F) v
9   | FV_All : forall F v, FV F (S v) -> FV (FAll F) v.

```

Definicija 6. Supstitucija je svaka funkcija iz skupa \mathbb{N} u skup svih terma. Supstituciju σ možemo promatrati kao niz terma t_0, t_1, t_2, \dots . Tada je *pomaknuta supstitucija*, s oznakom $t \cdot \sigma$, supstitucija koja odgovara nizu t, t_0, t_1, t_2, \dots , za neki term t .

Domena supstitucije može se rekurzivno proširiti na skup svih terma i skup svih formula.

```

1 Fixpoint subst_term (σ : var -> term) (t : term) : term :=
2   match t with
3   | var_term v => σ v
4   | TFunc f args => TFunc f (V.map (subst_term σ) args)
5   end.

```

```

1 Fixpoint subst_formula
2   (σ : var -> term Σ) (φ : formula)
3   : formula :=
4   match φ return formula with
5   | FPred P args => FPred P (V.map (subst_term σ) args)
6   | FIndPred P args => FIndPred P (V.map (subst_term σ) args)
7   | FNeg ψ => FNeg (subst_formula σ ψ)
8   | FImp ψ ξ => FImp (subst_formula σ ψ) (subst_formula σ ξ)
9   | FAll ψ => FAll (subst_formula (up_term_term σ) ψ)
10  end.

```

Ovdje funkcija `up_term_term` brine da supstitucija σ mijenja samo one varijable koje nisu vezane. Pišemo $\varphi[\sigma]$ za primjenu supstitucije σ na formulu φ . Često korištene supstitucije formula su supstitucija varijable x termom t u formuli φ , s oznakom $\varphi[t/x]$, te supstitucija svake varijable n u formuli φ varijablom $n + 1$, s oznakom φ^\dagger . Iste notacije

koristimo i za supstitucije na termima, listama terma i listama formula.

Konačno, potrebno je definirati sintaksu za indukciju. U Coqu su definicije induktivnih propozicija proizvoljne do na ograničenje pozitivnosti, no radi jednostavnosti u FOL_{ID} su moguće samo induktivne definicije s atomarnim formulama, a pišemo ih u stilu prirodne dedukcije:

$$\frac{Q_1 \mathbf{u}_1 \dots Q_n \mathbf{u}_n \quad P_1 \mathbf{v}_1 \dots P_m \mathbf{v}_m}{P \mathbf{t}}$$

Ovdje su Q_1, \dots, Q_n obični predikatni simboli, P_1, \dots, P_m i P su induktivni predikatni simboli, a podebljani znakovi predstavljaju n -torke terma, gdje je n mjesnost odgovarajućeg predikata.

Definicija 7. Produkcija je uređena četvorka

1. liste parova običnih predikatnih simbola i n -torki terma odgovarajućih duljina,
2. liste parova induktivnih predikatnih simbola i n -torki terma odgovarajućih duljina,
3. induktivnog predikatnog simbola P mjesnosti m i
4. m -torke terma.

```

1 Record production :=
2   mkProd {
3     preds : list {P : PredS Σ & vec (term Σ) (pred_ar P)};
4     indpreds : list {P : IndPredS Σ & vec (term Σ) (indpred_ar P)};
5     indcons : IndPredS Σ;
6     indargs : vec (term Σ) (indpred_ar indcons);
7   }.

```

Prvi i drugi član četvorke zovemo *premisama*, a treći i četvrti *konkluzijom*. U ostatku rada odabiremo neki podskup skupa svih produkcija koji zovemo *skupom induktivnih definicija*, a označavamo s Φ .

```

1 Definition IndDefSet := production -> Prop.

```

U idućim primjerima definiramo produkcije za induktivne predikatne simbole u proširenoj Peanovoj signaturi.

Primjer 3. Za broj kažemo da je prirodan ako je 0 ili je sljedbenik nekog prirodnog broja.

$$\frac{}{Nat(o)} (PA_prod_N_zero)$$

```

1 Definition PA_prod_N_zero : @production Σ__PA.
2   refine (@mkProd Σ__PA nil nil PA_Nat _).
3   refine [(@TFunc Σ__PA PA_zero V.nil)].
4 Defined.

```

$$\frac{Nat(x)}{Nat(s(x))} (PA_prod_N_succ)$$

```

1 Definition PA_prod_N_succ : @production Σ__PA.
2   refine (@mkProd Σ__PA nil _ PA_Nat _).
3   - refine (cons _ nil). exists PA_Nat; refine [(var_term 0)].
4   - refine [(@TFunc Σ__PA PA_succ ([var_term 0]))].
5 Defined.

```

Napomena. U primjeru 3 vidimo alternativni način definiranja u Coqu. Korisnik umjesto definiranja terma u Gallini ulazi u način dokazivanja te naredbom `refine` postepeno konstruira („profinjuje”) rezultatni term.

Primjer 4. Za broj kažemo da je paran ako je 0 ili je sljedbenik nekog neparnog broja.

$$\frac{}{Even(o)} (PA_prod_E_zero)$$

```

1 Definition PA_prod_E_zero : @production Σ__PA.
2   refine (@mkProd Σ__PA nil nil PA_Even _).
3   refine ( [ @TFunc Σ__PA PA_zero ([]) ] ).
4 Defined.

```

$$\frac{Odd(x)}{Even(s(x))} (PA_prod_E_succ)$$

```

1 Definition PA_prod_E_succ : @production Σ__PA.
2   refine (@mkProd Σ__PA nil _ PA_Even _).
3   - refine (cons _ nil). exists PA_Odd; refine ([var_term 0]).
4   - refine [_].
5   refine (@TFunc Σ__PA PA_succ _).
6   refine [var_term 0].
7 Defined.

```

Za broj kažemo da je neparan ako je sljedbenik nekog parnog broja.

$$\frac{Even(x)}{Odd(s(x))} (PA_prod_O_succ)$$

```

1 Definition PA_prod_0_succ : @production Σ__PA.
2   refine (@mkProd Σ__PA nil _ PA_Odd _).
3   - refine (cons _ nil). exists PA_Even; refine ([var_term 0]).
4   - refine [_].
5     refine (@TFunc Σ__PA PA_succ _).
6     refine ([var_term 0]).
7 Defined.

```

Skup prethodnih pet produkcija nazivamo skupom induktivnih definicija za proširenu Peanovu signaturu, s oznakom Φ_{PA} .

```

1 Inductive Φ__PA : @production Σ__PA -> Prop :=
2 | ID_N_zero : Φ__PA PA_prod_N_zero
3 | ID_N_succ : Φ__PA PA_prod_N_succ
4 | ID_E_zero : Φ__PA PA_prod_E_zero
5 | ID_E_succ : Φ__PA PA_prod_E_succ
6 | ID_0_succ : Φ__PA PA_prod_0_succ.

```

3.2. Semantika

Definicija 8. *Struktura prvog reda* (kratko: struktura) je uređena četvorka skupa M koji nazivamo *nosačem* te interpretacija funkcijskih, običnih predikatnih i induktivnih predikatnih simbola. Funkcijski simboli mjesnosti n interpretiraju se kao n -mjesne funkcije, a predikatni simboli mjesnosti n kao n -mjesne relacije na nosaču. Koristit ćemo ime nosača kao sinonim za čitavu strukturu, a interpretacije označavati s f^M odnosno P^M .

```

1 Structure structure := {
2   domain :> Set;
3   interpF (f : FuncS Σ) : vec domain (fun_ar f) -> domain;
4   interpP (P : PredS Σ) : vec domain (pred_ar P) -> Prop;
5   interpIP (P : IndPredS Σ) : vec domain (indpred_ar P) -> Prop;
6 }.

```

Primjer 5. Prije no što možemo definirati strukturu za Peanovu aritmetiku s induktivnim predikatima, odnosno strukturu za signaturu Σ_{PA} i skup induktivnih definicija Φ_{PA} , potrebno je definirati odgovarajuće interpretacije induktivnih predikatnih simbola u Coqu, u skladu s primjerima 3 i 4

```

1 Inductive NAT : nat -> Prop :=
2 | NO : NAT 0
3 | NS : forall n, NAT n -> NAT (S n).
4
5 Inductive EVEN : nat -> Prop :=
6 | EO : EVEN 0
7 | ES : forall n, ODD n -> EVEN (S n)
8 with ODD : nat -> Prop :=
9 | OS : forall n, EVEN n -> ODD (S n).

```

Ovdje su predikati EVEN i ODD definirani mehanizmom simultane rekurzije.

Konačno možemo definirati strukturu M_{PA} s nosačem \mathbb{N} te uobičajenom interpretacijom funkcijskih i običnih predikatnih simbola. Induktivne predikatne simbole *Nat*, *Even* i *Odd* interpretiramo redom Coqovim predikatima NAT, EVEN i ODD.

```

1 Definition M_PA : @structure Σ_PA.
2   refine (Build_structure nat _ _ _).
3   - intros []; cbn.
4     + intros. exact 0.
5     + intros n. exact (S (V.hd n)).
6     + intros xy. exact (V.hd xy + V.hd (V.tl xy)).
7     + intros xy. exact (V.hd xy * V.hd (V.tl xy)).
8   - intros []. cbn. intros args.
9     exact (V.hd args = V.hd (V.tl args)).
10  - intros []; cbn; intros args.
11    + exact (NAT (V.hd args)).
12    + exact (EVEN (V.hd args)).
13    + exact (ODD (V.hd args)).
14 Defined.

```

Definicija 9. Neka je M proizvoljna struktura. *Okolina* ρ za M je proizvoljna funkcija iz skupa prirodnih brojeva u nosač strukture.

```

1 Definition env := var -> M.

```

Okolina se može interpretirati kao niz d_0, d_1, d_2, \dots . Tada je *pomaknuta okolina*, s oznakom $d \cdot \rho$, niz d, d_0, d_1, d_2, \dots za neki $d \in M$. Proširenje domene okoline ρ na skup svih terma zovemo *evaluacijom*.


```

1 Fixpoint eval (ρ : env) (t : term Σ) : M :=
2   match t with
3   | var_term x => ρ x
4   | TFunc f args => interpF f (V.map (eval ρ) args)
5   end.

```

Pišemo t^ρ za evaluaciju terma t u okolini ρ . Istu notaciju koristimo i za evaluaciju n -torki terma.

Definicija 10. Neka je M proizvoljna struktura te ρ okolina za M . Istinitost formule φ u okolini ρ , s oznakom $\rho \models \varphi$, definiramo rekurzivno na način:

1. ako je P (obični ili induktivni) predikatni simbol mjesnosti n te su u_1, \dots, u_n termi, onda vrijedi $\rho \models P(u_1, \dots, u_n)$ ako i samo ako vrijedi $P^M(\rho(u_1), \dots, \rho(u_n))$,
2. vrijedi $\rho \models \neg\varphi$ ako i samo ako ne vrijedi $\rho \models \varphi$ (što još pišemo $\rho \not\models \varphi$),
3. vrijedi $\rho \models \varphi \rightarrow \psi$ ako i samo ako vrijedi $\rho \not\models \varphi$ ili $\rho \models \psi$ i
4. vrijedi $\rho \models \forall\varphi$ ako i samo ako za sve $d \in M$ vrijedi $d \cdot \rho \models \varphi$

```

1 Fixpoint Sat (ρ : env M) (F : formula Σ) : Prop :=
2   match F with
3   | FPred P args => interpP P (V.map (eval ρ) args)
4   | FIndPred P args => interpIP P (V.map (eval ρ) args)
5   | FNeg G => ~ Sat ρ G
6   | FImp F G => Sat ρ F -> Sat ρ G
7   | FAll G => forall d, Sat (d .: ρ) G
8   end.

```

Primjer 6. Formula $\forall x, N(x) \rightarrow E(x) \vee O(x)$ je istinita u strukturi M_{PA} bez obzira na njenu okolinu.

```

1 Lemma every_nat_is_even_or_odd_Sat :
2   forall (ρ : env M_PA), ρ ⊨ every_nat_is_even_or_odd.

```

Lema 1. Neka su φ, σ, M i ρ redom proizvoljna formula, supstitucija, struktura i okolina za M . Tada vrijedi $\rho \models \varphi[\sigma]$ ako i samo ako vrijedi $(t \mapsto t^\rho) \circ \sigma \models \varphi$.

```

1 Lemma strong_form_subst_sanity2 :
2   forall (φ : formula Σ) (σ : var -> term Σ)
3     (M : structure Σ) (ρ : env M),
4     ρ ⊨ (subst_formula σ φ) <-> (σ >> eval ρ) ⊨ φ.

```

Kompoziciju supstitucije σ i evaluacije $t \mapsto t^\rho$ možemo nazvati *semantičkom* supstitu-

cijom jer prvo provodi *sintaktičku* supstituciju σ nakon čega provodi evaluaciju. Tada možemo neformalno reći da sintaktička i semantička supstitucija komutiraju pod relacijom istinitosti.

3.3. Standardni modeli

Želimo ograničiti semantička razmatranja na samo one strukture koje „imaju smisla” za induktivne predikate. Prisjetimo se, predikatni simbol P mjesnosti n interpretira se na strukturi M podskupom skupa M^n . Indukciju smatramo dokazivanjem u razinama pa ima smisla promatrati *razine interpretacije* induktivnog predikata, gdje je nulta razina prazan skup, a svaku iduću razinu konstruiramo pomoću produkcija induktivnog skupa definicija i prethodnih razina. Tako je prva razina onaj podskup kojeg možemo dobiti najviše jednom „primjenom produkcija”, druga je razina onaj podskup kojeg možemo dobiti pomoću najviše dviju primjena produkcija, i tako dalje. Na taj se način, korak po korak, gradi *smisljena* interpretacija induktivnih predikata. Napominjemo da se zbog mogućih međuovisnosti induktivnih predikata razine interpretacije definiraju simultano. Ovaj odjeljak posvećujemo formalizaciji ovih pojmova.

Definicija 11. Neka je M proizvoljna struktura te neka je pr proizvoljna produkcija induktivnog skupa definicija Φ , primjerice:

$$\frac{Q_1 \mathbf{u}_1 \dots Q_n \mathbf{u}_n \quad P_1 \mathbf{v}_1 \dots P_m \mathbf{v}_m}{P \mathbf{t}}$$

Neka je f proizvoljna interpretacija induktivnih predikatnih simbola. Tada definiramo $\varphi_{pr}(f)$ kao skup svih $|P|$ -torki \mathbf{d} elemenata nosača M za koje postoji okolina ρ za M takva da:

- za sve $i \in \{1, \dots, n\}$ vrijedi $\mathbf{u}_i^\rho \in Q_i^M$,
- za sve $j \in \{1, \dots, m\}$ vrijedi $\mathbf{v}_j^\rho \in f(P_j)$ i
- $\mathbf{d} = \mathbf{t}^\rho$.

```

1 Definition  $\varphi_{pr}$ 
2   (pr : production)
3   (args : forall P : IndPredS  $\Sigma$ , vec D (indpred_ar P) -> Prop)
4   (ds : vec D (indpred_ar (indcons pr)))
5   : Prop :=
6     exists ( $\rho$  : env M),
7     (forall Q us, List.In (Q; us) (preds pr) ->
8       interpP Q (V.map (eval  $\rho$ ) us)) /\
9     (forall P ts, List.In (P; ts) (indpreds pr) ->
10      args P (V.map (eval  $\rho$ ) ts)) /\
11      ds = V.map (eval  $\rho$ ) (indargs pr).

```

Operator φ_{pr} je formalizacija ideje primjene produkcije. Nadalje, potrebno je definirati operator koji će uzeti u obzir sve produkcije koje se odnose na P . Definiramo $\varphi_P(f)$ kao uniju svih $\varphi_{pr'}(f)$ gdje je pr' produkcija u kojoj se P javlja u konkluziji.

```

1 Definition  $\varphi_P$ 
2   (P : IndPredS  $\Sigma$ )
3   (args : forall P : IndPredS  $\Sigma$ , vec D (indpred_ar P) -> Prop)
4   : vec D (indpred_ar P) -> Prop.
5   refine (fun ds => _).
6   refine (@ex production (fun pr => _)).
7   refine (@ex (P = indcons pr /\  $\Phi$  pr) (fun '(conj Heq H $\Phi$ ) => _)).
8   rewrite Heq in ds.
9   exact ( $\varphi_{pr}$  pr args ds).
10 Defined.

```

Konačno, definiramo operator skupa definicija φ_Φ kao preslikavanje koje svakom induktivnom predikatnom simbolu P pridružuje skup $\varphi_P(f)$.

```

1 Definition  $\varphi_\Phi$ 
2   (args : forall P : IndPredS  $\Sigma$ , vec D (indpred_ar P) -> Prop)
3   : forall P : IndPredS  $\Sigma$ , vec D (indpred_ar P) -> Prop :=
4     fun P =>  $\varphi_P$  P args.

```

Operator φ_Φ omogućuje simultanu primjenu produkcija.

Napomena. Kako je funkcija f bila uvedena na samom početku prethodne definicije, u stvari definicija operatora φ_Φ glasi $\varphi_\Phi(f)(P) := \varphi_P(f)$.

Primjer 7. Ilustrirajmo operator φ_Φ na strukturi M_{PA} i skupu induktivnih definicija Φ_{PA} . Prisjetimo se, u proširenoj Peanovoj signaturi imamo tri induktivna predikatna simbola; Nat , $Even$ i Odd , svaki od kojih je jednomjesan. Dakle, domena i kodomena operatora

φ_Φ u ovom primjeru je skup $\mathcal{P}(\mathbb{N})^3$, gdje prva, druga i treća projekcija odgovaraju redom interpretacijama predikata Nat , $Even$ i Odd . Nadalje, vrijedi:

$$\varphi_{\Phi_{PA}, M_{PA}}(N, E, O) = (\{0\} \cup \{n + 1 \mid n \in N\}, \{0\} \cup \{o + 1 \mid o \in O\}, \{e + 1 \mid e \in E\}).$$

Prema notaciji u prethodnoj napomeni, funkcija f je ovdje trojka (N, E, O) , a indeksiranje induktivnim predikatnim simbolom odgovara navedenim projekcijama.

Propozicija 1. Operator φ_Φ je monoton.

```

1 Proposition  $\varphi\_Phi\_monotone$  :
2   forall (f g : forall P, vec D (indpred_ar P) -> Prop),
3     (forall P v, f P v -> g P v) ->
4     (forall P v,  $\varphi\_Phi$  f P v ->  $\varphi\_Phi$  g P v).

```

Definicija 12. Neka je M proizvoljna struktura. Definiramo aproksimaciju skupa induktivnih definicija Φ razine $\alpha \in \mathbb{N}$, u oznaci φ_Φ^α , rekursivno na način:

1. $\varphi_\Phi^0(P) := \emptyset$
2. $\varphi_\Phi^{\alpha+1} := \varphi_\Phi(\varphi_\Phi^\alpha)$.

```

1 Fixpoint  $\varphi\_Phi\_n$  P ( $\alpha$  : nat) (v : vec M (indpred_ar P)) : Prop :=
2   match  $\alpha$  with
3   | 0 => False
4   | S  $\alpha$  => @ $\varphi\_Phi$   $\Sigma$  M  $\Phi$  (fun P =>  $\varphi\_Phi\_n$  P  $\alpha$ ) P v
5   end.

```

Tada je *aproksimant* induktivnog predikatnog simbola P razine α upravo $\varphi_\Phi^\alpha(P)$.

```

1 Definition approximant_of (P : IndPredS  $\Sigma$ )
2   : nat -> vec M (indpred_ar P) -> Prop :=
3    $\varphi\_Phi\_n$  P.

```

Napomena. Brotherston je definirao aproksimaciju razine α za pojedini induktivni predikatni simbol kao uniju aproksimacija svih nižih razina. Takva je definicija ekvivalentna našoj.

Lema 2. Za svaki prirodni broj α i induktivni predikatni simbol P vrijedi

$$\varphi_\Phi^\alpha(P) = \bigcup_{\beta < \alpha} \varphi_\Phi^\beta(P).$$

```

1 Lemma approximant_characterization : forall α P v,
2   φ_Φ_n P α v <-> exists β,
3     β < α /\ @φ_Φ Σ M Φ (fun P => φ_Φ_n P β) P v.

```

Primjer 8. Za svaki prirodan broj α , a u skladu s primjerom 7, aproksimant razine α induktivnog predikatnog simbola Nat je skup svih prirodnih brojeva strogo manjih od α , odnosno vrijedi:

$$\varphi_{\Phi_{PA}, M_{PA}}^{\alpha}(Nat) = \{0, 1, \dots, \alpha - 1\}.$$

```

1 Lemma approximants_of_PA_Nat : forall α n,
2   @approximant_of Σ__PA M__PA Φ__PA PA_Nat α ([n])
3   <-> n < α.

```

Definicija 13. Aproksimacija razine ω , u oznaci φ_{Φ}^{ω} , je za svaki pojedini predikatni simbol unija aproksimacija razina manjih od ω .

```

1 Definition φ_Φ_ω P v := exists α, φ_Φ_n P α v.

```

Primjer 9. Aproksimant razine ω induktivnog predikatnog simbola Nat je cijeli skup prirodnih brojeva \mathbb{N} .

```

1 Lemma NAT_φ_Φ_ω: forall (n : nat), @φ_Φ_ω Σ__PA M__PA Φ__PA PA_Nat ([n]).

```

Lema 3. Aproksimacija razine ω je najmanji skup sa svojstvom $\varphi_{\Phi}(\varphi_{\Phi}^{\omega}) \subseteq \varphi_{\Phi}^{\omega}$.

```

1 Lemma ω_prefixed : forall P v, @φ_Φ Σ M Φ φ_Φ_ω P v -> φ_Φ_ω P v.
2 Lemma ω_least : forall args,
3   (forall P v, @φ_Φ Σ M Φ args P v -> args P v) ->
4   forall P v, φ_Φ_ω P v -> args P v.

```

Definicija 14. Kažemo da je struktura M *standardni model* za Φ ako interpretira svaki induktivni predikatni simbol njegovim aproksimantom razine ω .

```

1 Definition standard_model
2   (Σ : signature) (Φ: @IndDefSet Σ) (M : structure Σ) : Prop :=
3   forall (P : IndPredS Σ) ts, interpIP P ts <-> @φ_Φ_ω Σ M Φ P ts.

```

Prema lemi 3, standardni model je najmanja struktura koja ima smisla za zadani skup induktivnih definicija.

Propozicija 2. Struktura M_{PA} je standardni model za skup induktivnih definicija Φ_{PA} .

```
1 Lemma standard_model__PA : @standard_model  $\Sigma$ __PA  $\Phi$ __PA M__PA.
```

3.4. Sistem sekvenata s induktivnim definicijama

Cilj je ovog odjeljka definirati dokazni sustav *LKID* za logiku *FOL_{ID}*. Ovaj dokazni sustav temelji se na Gentzenovu računu sekvenata, a proširen je lijevim i desnim pravilima za indukciju. Prije no što definiramo *LKID*, potrebno je definirati pojam sekvente i međusobne zavisnosti induktivnih predikata.

Definicija 15. Neka su Γ i Δ proizvoljne liste formula. Tada je *sekventa* uređeni par (Γ, Δ) , a označavamo ju s $\Gamma \vdash \Delta$.

```
1 Inductive sequent : Set :=
2 | mkSeq ( $\Gamma$   $\Delta$  : list (formula  $\Sigma$ )).
```

Sekventom $\Gamma \vdash \Delta$ konceptualno tvrdimo da istinitost *svake* tvrdnje u Γ povlači istinitost *neke* tvrdnje u Δ . Kažemo da je sekventa $\Gamma \vdash \Delta$ *dokaziva*, ili da iz Γ postoji izvod za Δ , ako u sustavu *LKID* postoji *dokaz* za $\Gamma \vdash \Delta$. Pojam dokaza ćemo precizirati kasnije.

Definicija 16. Neka su P_i i P_j proizvoljni induktivni predikatni simboli. Kažemo da je simbol P_i u relaciji *Prem* sa simbolom P_j ako postoji produkcija u skupu induktivnih definicija Φ takva da se simbol P_i javlja u njenoj konkluziji te se simbol P_j javlja u njenim premisama.

```
1 Definition Prem (Pi Pj : IndPredS  $\Sigma$ ) :=
2   exists pr,  $\Phi$  pr /\
3     indcons pr = Pi /\
4     exists ts, List.In (Pj; ts) (indpreds pr).
```

Definicija 17. Definiramo relaciju *Prem** kao refleksivno i tranzitivno zatvorenje relacije *Prem* na skupu induktivnih predikatnih simbola.

```
1 Definition Prem_star := clos_refl_trans (IndPredS  $\Sigma$ ) Prem.
```

Za induktivne predikatne simbole P i Q kažemo da su *međusobno zavisni* ako vrijedi $Prem^*(P, Q)$ i $Prem^*(Q, P)$.

```
1 Definition mutually_dependent (P Q : IndredS  $\Sigma$ ) :=
2   Prem_star P Q /\ Prem_star Q P.
```

Za međusobno zavisne induktivne predikatne simbole kažemo još da su definirani simultanom rekurzijom. Coqovi predikati `EVEN` i `ODD` iz primjera 5 definirani su simultanom rekurzijom.

Primjer 10. U skupu induktivnih definicija Φ_{PA} , induktivni predikatni simboli *Even* i *Odd* su međusobno zavisni.

```
1 Example mut_dep_E_O :
2   @mutually_dependent  $\Sigma$ __PA  $\Phi$ __PA PA_Even PA_Odd.
```

S druge strane, induktivni predikatni simboli *Nat* i *Even* nisu međusobno zavisni.

```
1 Example not_mut_dep_N_E :
2   ~ @mutually_dependent  $\Sigma$ __PA  $\Phi$ __PA PA_Nat PA_Even.
```

Lema 4. Međusobna zavisnost je relacija ekvivalencije na skupu induktivnih predikatnih simbola.

```
1 Lemma mutually_dependent_equiv : equiv (IndPredS  $\Sigma$ ) mutually_dependent.
```

Sustav *LKID* sastoji se od četiri vrste pravila izvoda: strukturalna, propozicijska, pravila za kvantifikatore i pravila za induktivne definicije. Navest ćemo ih tim redom, a definirati kroz više blokova Coq koda. Napominjemo da se sustav *LKID* razlikuje od Gentzenova sustava *LK* na nekoliko mjesta, a zbog implementacijskih detalja te induktivnih definicija. Za neko pravilo izvoda kažemo da je *dopustivo* ako nije u definiciji sustava *LKID*, ali je iz originalnih pravila izvedivo.

```
1 Inductive LKID : sequent -> Prop :=
```

Strukturalna su **pravila** prikazana na slici 3.1., a služe baratanju strukturom sekvente. Iako je sekventa implementirana kao par listi formula, te liste se ponašaju kao skupovi u strukturalnim pravilima jer za provjeru pripadnosti koristimo Coqov predikat `In`. Posljedično, pravilo slabljenja *Wk* sa sobom povlači pravila permutacije i kontrakcije.

$$\begin{array}{c}
\frac{\Gamma \cap \Delta \neq \emptyset}{\Gamma \vdash \Delta} (Ax) \\
\frac{\Gamma' \vdash \Delta' \quad \Gamma' \subseteq \Gamma \quad \Delta' \subseteq \Delta}{\Gamma \subseteq \Delta} (Wk) \\
\frac{\Gamma \vdash \varphi, \Delta \quad \varphi, \Gamma \vdash \Delta}{\Gamma \vdash \Delta} (Cut) \\
\frac{\Gamma \vdash \Delta}{\Gamma[\sigma] \vdash \Delta[\sigma]} (Subst)
\end{array}$$

Slika 3.1. Strukturalna pravila sustava *LKID*.

```

2  (* Structural rules. *)
3  | Ax : forall Γ Δ φ, In φ Γ -> In φ Δ -> LKID (Γ ⊢ Δ)
4  | Wk : forall Γ' Δ' Γ Δ,
5      Γ' ⊆ Γ ->
6      Δ' ⊆ Δ ->
7      LKID (Γ' ⊢ Δ') ->
8      LKID (Γ ⊢ Δ)
9  | Cut : forall Γ Δ φ,
10     LKID (Γ ⊢ φ :: Δ) ->
11     LKID (φ :: Γ ⊢ Δ) ->
12     LKID (Γ ⊢ Δ)
13 | Subst : forall Γ Δ,
14     LKID (Γ ⊢ Δ) ->
15     forall σ, LKID (map (subst_formula σ) Γ ⊢ map (subst_formula σ) Δ)

```

Primjer 11. Prošireno pravilo aksioma je dopustivo.

$$\frac{}{\varphi, \Gamma \vdash \varphi, \Delta} (AxExtended)$$

Lemma AxExtended : forall Γ Δ φ, LKID (φ :: Γ ⊢ φ :: Δ).

Pravilo permutacije je dopustivo.

$$\frac{\Gamma' \vdash \Delta' \quad \Gamma' \sim \Gamma \quad \Delta' \sim \Delta}{\Gamma \vdash \Delta} (Perm)$$

Ovdje znakom \sim označujemo relaciju permutacije na listama.

Lemma Perm : forall Γ' Δ' Γ Δ,
 Permutation Γ' Γ ->
 Permutation Δ' Δ ->
 LKID (Γ' ⊢ Δ') ->
 LKID (Γ ⊢ Δ).

Propozicijska su pravila prikazana na slici 3.2., a služe rasuđivanju s implikacijama i negacijama. Ova su pravila identična u sustavima *LK* i *LKID*.

$$\frac{\Gamma \vdash \varphi, \Delta}{\neg \varphi, \Gamma \vdash \Delta} (NegL)$$

$$\frac{\varphi, \Gamma \vdash \Delta}{\Gamma \vdash \neg \varphi, \Delta} (NegR)$$

$$\frac{\Gamma \vdash \varphi, \Delta \quad \psi, \Gamma \vdash \Delta}{\varphi \rightarrow \psi, \Gamma \vdash \Delta} (Impl)$$

$$\frac{\varphi, \Gamma \vdash \psi, \Delta}{\Gamma \vdash \varphi \rightarrow \psi, \Delta} (ImpR)$$

Slika 3.2. Propozicijska pravila sustava *LKID*.

```

16 (* Propositional rules. *)
17 | NegL : forall Γ Δ φ, LKID (Γ ⊢ φ :: Δ) -> LKID (FNeg φ :: Γ ⊢ Δ)
18 | NegR : forall Γ Δ φ, LKID (φ :: Γ ⊢ Δ) -> LKID (Γ ⊢ FNeg φ :: Δ)
19 | Impl : forall Γ Δ φ ψ,
20     LKID (Γ ⊢ φ :: Δ) -> LKID (ψ :: Γ ⊢ Δ) ->
21     LKID (FImp φ ψ :: Γ ⊢ Δ)
22 | ImpR : forall Γ Δ φ ψ,
23     LKID (φ :: Γ ⊢ ψ :: Δ) -> LKID (Γ ⊢ (FImp φ ψ) :: Δ)

```

Primjer 12. Desno pravilo za konjunkciju je dopustivo.

$$\frac{\Gamma \vdash \varphi, \Delta \quad \Gamma \vdash \psi, \Delta}{\Gamma \vdash \varphi \wedge \psi, \Delta} (AndR)$$

```

Lemma AndR : forall Γ Δ φ ψ,
  LKID (Γ ⊢ φ :: Δ) -> LKID (Γ ⊢ ψ :: Δ) ->
  LKID (Γ ⊢ FAnd φ ψ :: Δ).

```

Lijevo pravilo za disjunkciju je dopustivo.

$$\frac{\varphi, \Gamma \vdash \Delta \quad \psi, \Gamma \vdash \Delta}{\varphi \vee \psi, \Gamma \vdash \Delta} (OrL)$$

```

Lemma OrL : forall Γ Δ φ ψ,
  LKID (φ :: Γ ⊢ Δ) -> LKID (ψ :: Γ ⊢ Δ) ->
  LKID (FOr φ ψ :: Γ ⊢ Δ).

```

Pravila za kvantifikatore prikazana su na slici 3.3., a odnose se na univerzalnu kvantifikaciju. Ova se pravila razlikuju od analognih u sustavu *LK* zbog korištenja de Bruijnovih indeksa kod kvantifikacije. U lijevom pravilu koristimo supstituciju $t \cdot \sigma_{id}$

u formuli φ , gdje je σ_{id} supstitucija identiteta, kako bismo iskazali da se term t javlja na mjestu neke varijable x . Za desno pravilo, primijetimo da se varijabla 0 ne javlja u listama formula Γ^\uparrow i Δ^\uparrow zbog pomicanja. Kako implicitno kvantificiramo po varijabli 0, a ona se javlja (eventualno) samo u formuli φ , smijemo univerzalno kvantificirati formulu φ .

$$\frac{\varphi[t \cdot \sigma_{id}], \Gamma \vdash \Delta}{\forall \varphi, \Gamma \vdash \Delta} (AllL)$$

$$\frac{\Gamma^\uparrow \vdash \varphi, \Delta^\uparrow}{\Gamma \vdash \forall \varphi, \Delta} (AllR)$$

Slika 3.3. Pravila za kvantifikatore sustava *LKID*.

```

24 (* Quantifier rules. *)
25 | AllL : forall Γ Δ φ t,
26   LKID (subst_formula (t :: ids) φ :: Γ ⊢ Δ) ->
27   LKID (FAll φ :: Γ ⊢ Δ)
28 | AllR : forall Γ Δ φ,
29   LKID (shift_formulas Γ ⊢ φ :: shift_formulas Δ) ->
30   LKID (Γ ⊢ (FAll φ) :: Δ)

```

Primjer 13. Lijevo pravilo za egzistencijalnu kvantifikaciju je dopustivo.

$$\frac{\varphi, \Gamma^\uparrow \vdash \Delta^\uparrow}{\exists \varphi, \Gamma \vdash \Delta} (ExistL)$$

```

Lemma ExistL : forall Γ Δ φ,
  LKID (φ :: shift_formulas Γ ⊢ shift_formulas Δ) ->
  LKID (FExist φ :: Γ ⊢ Δ).

```

```

31 | IndL : forall  $\Gamma$   $\Delta$  (Pj : IndPredS  $\Sigma$ ) (u : vec (term  $\Sigma$ ) (indpred_ar Pj))
32     (z_i : forall P, vec var (indpred_ar P)) (* dodati pretpostavku forall
33     (G_i : IndPredS  $\Sigma$  -> formula  $\Sigma$ )
34     (HG2 : forall Pi, ~mutually_dependent Pi Pj -> G_i Pi = FIndPred Pi (V
35 let max $\Gamma$  := max_fold (map some_var_not_in_formula  $\Gamma$ ) in
36 let max $\Delta$  := max_fold (map some_var_not_in_formula  $\Delta$ ) in
37 let maxP := some_var_not_in_formula (FIndPred Pj u) in
38 let shift_factor := max maxP (max max $\Gamma$  max $\Delta$ ) in
39 let Fj := subst_formula (finite_subst (z_i Pj) u) (G_i Pj) in
40 let minor_premises :=
41     (forall pr (Hdep : mutually_dependent (indcons pr) Pj),
42      let Qs := shift_formulas_by shift_factor (FPreds_from_preds (preds pr))
43      let Gs := map (fun '(P; args) =>
44          let shifted_args := V.map (shift_term_by shift_factor)
45          let  $\sigma$  := finite_subst (z_i P) (shifted_args) in
46          let G := G_i P in
47          subst_formula  $\sigma$  G)
48          (indpreds pr) in
49      let Pi := indcons pr in
50      let ty := V.map (shift_term_by shift_factor) (indargs pr) in
51      let Fi := subst_formula (finite_subst (z_i Pi) ty) (G_i Pi) in
52      LKID (Qs ++ Gs ++  $\Gamma \vdash Fi :: \Delta$ ))
53 in
54 minor_premises ->
55 LKID (Fj ::  $\Gamma \vdash \Delta$ ) ->
56 LKID (FIndPred Pj u ::  $\Gamma \vdash \Delta$ )
57 | IndR : forall  $\Gamma$   $\Delta$  pr  $\sigma$ ,
58    $\Phi$  pr ->
59   (forall Q us, In (Q; us) (preds pr) ->
60     LKID ( $\Gamma \vdash$  (FPred Q (V.map (subst_term  $\sigma$ ) us) ::  $\Delta$ ))) ->
61   (forall P ts, In (P; ts) (indpreds pr) ->
62     LKID ( $\Gamma \vdash$  (FIndPred P (V.map (subst_term  $\sigma$ ) ts) ::  $\Delta$ ))) ->
63   LKID ( $\Gamma \vdash$  FIndPred (indcons pr) (V.map (subst_term  $\sigma$ ) (indargs pr)) ::  $\Delta$ ).

```

3.5. Adekvatnost

Lokalne adekvatnosti za pravila izvoda. Glavni teorem.

4. Ciklički dokazi

Koinduktivni tip podatka i koinduktivna propozicija. Jedan primjer su Streamovi i predikat `Infinite`. Jednostavniji primjer bi možda bio koinduktivni `nat` i koinduktivni `le`.

Kako bi izgledali ciklički dokazi u LKID? Ono što je tamo “repeat funkcija” je u Coqu `cofix`.

5. Zaključak

Literatura

- [1] Y. Bertot i P. Castéran, *Interactive theorem proving and program development: Coq'Art: the Calculus of Inductive Constructions*. Springer Science & Business Media, 2013.
- [2] B. C. Pierce, A. A. de Amorim, C. Casinghino, M. Gaboardi, M. Greenberg, C. Hrițcu, V. Sjöberg, i B. Yorgey, *Logical Foundations*, ser. Software Foundations, B. C. Pierce, Ur. Electronic textbook, 2023., sv. 1, version 6.5, <http://softwarefoundations.cis.upenn.edu>.
- [3] B. C. Pierce, A. A. de Amorim, C. Casinghino, M. Gaboardi, M. Greenberg, C. Hrițcu, V. Sjöberg, A. Tolmach, i B. Yorgey, *Programming Language Foundations*, ser. Software Foundations, B. C. Pierce, Ur. Electronic textbook, 2024., sv. 2, version 6.5, <http://softwarefoundations.cis.upenn.edu>.
- [4] A. W. Appel, *Verified Functional Algorithms*, ser. Software Foundations, B. C. Pierce, Ur. Electronic textbook, 2023., sv. 3, version 1.5.4, <http://softwarefoundations.cis.upenn.edu>.
- [5] A. Chlipala, *Certified programming with dependent types: a pragmatic introduction to the Coq proof assistant*. MIT Press, 2022.
- [6] The Coq Development Team, “The Coq Reference Manual, Release 8.18.0”, <https://coq.inria.fr/doc/v8.18/refman/>, 2023.
- [7] T. Coquand i G. Huet, “The calculus of constructions”, INRIA, teh. izv. RR-0530, svibanj 1986. [Mrežno]. Adresa: <https://inria.hal.science/inria-00076024>

- [8] F. Pfenning i C. Paulin-Mohring, “Inductively Defined Types in the Calculus of Constructions”, u *Proceedings of the 5th International Conference on Mathematical Foundations of Programming Semantics*. Berlin, Heidelberg: Springer-Verlag, 1989., str. 209–228.
- [9] E. Giménez, “Codifying guarded definitions with recursive schemes”, u *Types for Proofs and Programs*, P. Dybjer, B. Nordström, i J. Smith, Ur. Berlin, Heidelberg: Springer Berlin Heidelberg, 1995., str. 39–59.
- [10] M. Sozeau, S. Boulier, Y. Forster, N. Tabareau, i T. Winterhalter, “Coq Coq correct! verification of type checking and erasure for Coq, in Coq”, *Proceedings of the ACM on Programming Languages*, sv. 4, br. POPL, str. 1–28, 2019.
- [11] G. Hutton, *Programming in Haskell*, 2. izd. Cambridge University Press, 2016.
- [12] J. Brotherston, “Cyclic proofs for first-order logic with inductive definitions”, u *Automated Reasoning with Analytic Tableaux and Related Methods*, B. Beckert, Ur. Berlin, Heidelberg: Springer Berlin Heidelberg, 2005., str. 78–92.
- [13] —, “Sequent calculus proof systems for inductive definitions”, doktorska disertacija, School of Informatics, University of Edinburgh, 2006.
- [14] M. Vuković, *Matematička logika*. Element, 2009.
- [15] N. G. de Bruijn, “Lambda calculus notation with nameless dummies, a tool for automatic formula manipulation, with application to the Church-Rosser theorem”, *Indagationes Mathematicae (Proceedings)*, sv. 75, br. 5, str. 381–392, 1972.
- [16] B. C. Pierce, *Types and Programming Languages*. MIT Press, 2002.
- [17] K. Stark, “Mechanising Syntax with Binders in Coq”, doktorska disertacija, Saarland University, 2020.
- [18] K. Stark, S. Schäfer, i J. Kaiser, “Autosubst 2: Reasoning with Multi-Sorted de Bruijn Terms and Vector Substitutions”, *8th ACM SIGPLAN International Conference on Certified Programs and Proofs, CPP 2019, Cascais, Portugal, January 14-15, 2019*, 2019.

Sažetak

Primjene Coq alata za dokazivanje u matematici i računarstvu

Miho Hren

Unesite sažetak na hrvatskom.

Ključne riječi: prva ključna riječ; druga ključna riječ; treća ključna riječ

Abstract

Applications of the Coq Proof Assistant in mathematics and computer science

Miho Hren

Enter the abstract in English.

Keywords: the first keyword; the second keyword; the third keyword