

Ovo je zahvala.

Sadržaj

1. Uvod	2
2. Coq	3
2.1. Što je Coq?	3
2.2. Programiranje u Coqu	5
2.3. Hijerarhija tipova	8
2.4. Propozicije i tipovi, dokazi i termi	10
2.5. Ograničenja tipskog sustava	11
3. Logika prvog reda s induktivnim definicijama	14
3.1. Sintaksa	15
3.2. Semantika	19
3.3. Standardni modeli	20
3.4. Sistem sekvenata s induktivnim definicijama	24
3.5. Adekvatnost	24
4. Ciklički dokazi	25
5. Zaključak	26
Literatura	27
Sažetak	29
Abstract	30

1. Uvod

2. Coq

U ovom poglavlju dajemo pogled svisoka na programski sustav Coq. Prvo ćemo objasniti što je uopće Coq, u kojem je kontekstu nastao, i od kojih komponenti se sastoji. Zatim ćemo dati kratak pregled programiranja u Coqu, nakon čega ćemo se baviti naprednijim konceptima i spomenuti neka ograničenja. Za širi opseg gradiva, čitatelja upućujemo na knjige *Coq'Art* [1], *Software Foundations* [2, 3, 4] i *Certified Programming with Dependent Types* [5] te na službenu dokumentaciju [6].

2.1. Što je Coq?

Alat za dokazivanje Coq¹, punog naziva *The Coq Proof Assistant*, programski je sustav pomoću kojeg korisnici mogu dokazivati matematičke tvrdnje. **Misao:** *ne služi samo tome, može biti i općeniti funkcijski programski jezik, može služiti za programiranje sa zavisnim tipovima* Alat se temelji na λ -računu i teoriji tipova, a prva je inačica implementirana godine 1984. [6] Ovaj rad koristi inačicu 8.18 iz rujna godine 2023.

Program Coq može se pokrenuti u interaktivnom ili u skupnom načinu rada. Interaktivni način rada pokreće se naredbom `coqtop`, a korisniku omogućuje rad u ljusci sličnoj `bash` i `python` ljuskama. Interaktivna ljuska (također poznata pod imenom *toplevel*) služi unosu definicija i iskazivanju lema. Skupni način rada pokreće se naredbom `coqc`, a korisniku omogućuje semantičku provjeru i prevođenje izvornih datoteka u jednostavnije formate. Kod formaliziranja i dokazivanja, korisnik će najčešće koristiti interaktivni način rada, po mogućnosti kroz neku od dostupnih razvojnih okolina.²

Misao: *spomenuti proof mode? spomenuti workflow dokazivanja? ima jedna zgodna*

¹<https://coq.inria.fr/>

²Autor rada koristio je paket *Proof General* za uređivač teksta *Emacs*. Druge često korištene okoline su *VsCoq* i *CoqIDE*.

slika u QED at Large **Misao:** *treba naglasiti da je Coq interaktivni dokazivač teorema*

Kao programski jezik, Coq se sastoji od više podjezika različitih namjena, od kojih spominjemo *Vernacular*, *Gallinu* i *Ltac*.

Vernacular **Misao:** *vernacular znači „govorni jezik”* je jezik naredbi kojima korisnik komunicira sa sustavom (i u interaktivnom i u skupnom načinu rada); svaka Coq skripta (datoteka s nastavkom `.v`) je niz naredbi. Neke od najčešće korištenih naredbi su `Check`, `Definition`, `Inductive`, `Fixpoint` i `Lemma`. Pomoću naredbi za tvrdnje, kao što je `Lemma`, Coq prelazi iz *toplevela* **Misao:** *treba bolji prevod* u način dokazivanja (*proof mode*).

Gallina je Coqov strogo statički tipiziran specifikacijski jezik. Kako se glavnina programiranja u Coqu svodi upravo na programiranje u Gallini, posvećujemo joj idući odjeljak.

Ltac je Coqov netipiziran jezik za definiciju i korištenje taktika. Taktike su pomoćne naredbe kojima se u načinu dokazivanja konstruira dokaz. Može se reći da je *Ltac* jezik za metaprogramiranje Galline. Primjeri taktika su `intros`, `destruct`, `apply` i `rewrite`.

Pogledajmo ilustrativan primjer.

```
1  Lemma example_lemma : 1 + 1 = 2.  
2  Proof.  
3    cbn. reflexivity.  
4  Qed.
```

Ključne riječi `Lemma`, `Proof` i `Qed` dio su *Vernaculara*, izraz `example_lemma : 1 + 1 = 2` dio je *Galline*, a pomoćne naredbe `cbn` i `reflexivity` dio su *Ltaca*.

Jezgra programskog sustava Coq je algoritam za provjeru tipova (*type checking*) implementiran u OCamlu — svaka tvrdnja koja se dokazuje izrečena je pomoću tipova. Ostatak sustava u načelu služi za knjigovodstvo i poboljšanje korisničkog iskustva. Nužno je da jezgra sustava bude relativno mala kako bismo se mogli uvjeriti u njenu točnost. U suprotnom, možemo li biti sigurni da su naše dokazane tvrdnje doista istinite? **Misao:** *kažem istinite, ali u stvari mislim dokazive, no to je nespretno za napisati i diskusija oko toga je preopćenita*

Prve inačice Coqa implementirale su račun konstrukcija, no kasnije je dodana podr-

ška za induktivno i koinduktivno definirane tipove [7, 8]. Danas se može reći da Coq implementira polimorfni kumulativni račun induktivnih konstrukcija [9]. **Misao: mogu spomenuti i preteče računa konstrukcija i jezike koji ih implementiraju, npr. Lisp je implementacija λ -računa** Coq se, osim kao dokazivač teorema, može koristiti i za programiranje sa zavisnim tipovima. U toj sferi konkuriraju jezici Agda³, Idris⁴ i Lean⁵. Coq se između njih ističe po usmjerenosti prema dokazivanju, posebno po korištenju taktika (jezik Ltac) i nepredikativnoj sorti Prop (o kojoj će kasnije biti riječi). Još jedna prednost Coqa je mehanizam *ekstrakcije* pomoću kojeg korisnik može proizvoljnu funkciju⁶ prevesti u jezik niže razine apstrakcije.⁷ Mehanizam ekstrakcije nije dokazano točan, no poželjno je da funkcije zadržavaju točnost i nakon ekstrakcije pa se radi na verifikaciji ekstrakcije [9].

2.2. Programiranje u Coqu

Gallina je funkcijski programski jezik, što znači da su funkcije prvoklasni objekti — funkcije mogu biti argumenti i povratne vrijednosti drugih funkcija. Dodatno, varijable su nepromjenjive (*immutable*) te se iteracija ostvaruje rekurzijom. Za uvod u funkcijsko programiranje, čitatelja upućujemo na knjigu *Programming in Haskell* [10]. Primjeri koje ćemo vidjeti u ostatku ovog odjeljka dijelom se oslanjaju na tipove i funkcije definirane u Coqovoj standardnoj knjižnici.⁸

Gallina je strogo statički tipiziran jezik, što znači da se svakom termu prilikom prevođenja dodjeljuje tip⁹. Naredbom `Check` možemo provjeriti tip nekog terma ili doznati da se termu ne može dodijeliti tip. Dalje u radu pod „term” mislimo na dobro formirane terme, odnosno na one terme kojima se može dodijeliti tip. Kažemo da je term *stanovnik* tipa koji mu je dodijeljen. Za tip kažemo da je *nastanjen*, odnosno *nenastanjen*, ako postoji, odnosno ne postoji, stanovnik tog tipa. Kako su u Coqu i tipovi termi, radi razumljivosti i zvučnosti umjesto „tip tipa” kažemo „sorta tipa”.

³<https://wiki.portal.chalmers.se/agda/>

⁴<https://www.idris-lang.org/>

⁵<https://lean-lang.org/>

⁶Za koju je dokazao točnost, štogod to značilo.

⁷Trenutno su podržani Haskell, OCaml i Scheme.

⁸<https://coq.inria.fr/library/>

⁹Tipovi su kolekcije srodnih objekata.

Kao i u ostalim jezicima, kod programiranja u Coqu korisnik se oslanja na dostupne primitivne izraze, od kojih su najvažniji:

- `forall`, pomoću kojeg se konstruiraju funkcijski tipovi i zavisni produkti;
- `match`, pomoću kojeg se provodi *pattern matching*;
- `fun`, pomoću kojeg se definiraju funkcije;
- `fix`, pomoću kojeg se definiraju rekurzivne funkcije i
- `cofix`, pomoću kojeg se definiraju korekurzivne funkcije.

Jedna od osnovnih naredbi za stvaranje novih terma je naredba `Definition`.

```
1 Definition negb (b : bool) : bool :=
2 match b with
3 | false => true
4 | true  => false
5 end.
```

U gornjem kodu definirana je funkcija `negb` čiji se argument `b` tipa `bool` destrukture te se vraća njegova negacija, također tipa `bool`. Važno je napomenuti da svaki `match` izraz mora imati po jednu granu za svaki konstruktor tipa.¹⁰ U ovom su primjeru konstante `false` i `true` jedini konstruktori tipa `bool`. Funkcija `negb` je tipa `bool → bool`.

```
1 Definition mult_zero_r : Prop := forall (n : nat), n * 0 = 0.
```

Ovdje je definirana propozicija (tip) imena `mult_zero_r` kao univerzalno kvantificirana tvrdnja po prirodnim brojevima.

Naredbom `Inductive` definira se *induktivni* tip te se automatski za njega generiraju principi *indukcije* i *rekurzije*.

```
1 Inductive nat : Set :=
2 | 0 : nat
3 | S : nat -> nat.
```

Ovim kodom definirali smo tri terma:

- `nat` (tip prirodnih brojeva) je term sorte `Set`,
- `0` (broj nula) je term tipa `nat` i
- `S` (funkcija sljedbenika) je term tipa `nat → nat`.

Za term `nat` kažemo da je konstruktor tipa (*type constructor*), a za terme `0` i `S` kažemo da su konstruktori objekata (*object constructors*).

¹⁰U tandemu s uvjetom strukturalne rekurzije, ovime je osigurana totalnost svake funkcije.

Rekurzija nad induktivnim tipovima može se ostvariti naredbom `Fixpoint`, koja u pozadini koristi izraz `fix`.

```
1 Fixpoint plus (n m : nat) {struct n} : nat :=  
2 match n with  
3 | 0 => m  
4 | S n' => S (plus n' m)  
5 end.
```

U ovom primjeru definirana je funkcija `plus` koja prima dva argumenta tipa `nat`. Funkcija je rekurzivna po prvom argumentu što je vidljivo oznakom `{struct n}`. Napominjemo da su induktivni tipovi dobro utemeljeni, to jest svaki term induktivnog tipa je konačan.

Osim induktivnih, u Coqu postoje i koinduktivni tipovi, koji nisu dobro utemeljeni, zbog čega za njih nije moguće definirati principe indukcije i rekurzije. Umjesto rekurzije, koinduktivni tipovi koriste se u korekurzivnim funkcijama. Standardan primjer koinduktivnog tipa je beskonačna lista.

```
1 Set Primitive Projections.  
2 CoInductive Stream (A : Type) := {  
3     hd : A;  
4     tl : Stream A;  
5 }
```

Ovime smo definirali familiju tipova `Stream` indeksiranu tipskom varijablom `A`. Svaki `Stream` ima glavu i rep koji je također `Stream`.

Stanovnici koinduktivnih tipova konstruiraju se korekurzivnim funkcijama naredbom `CoFixpoint`, koja u pozadini koristi `cofix`.

```
1 Cofixpoint from (n : nat) : Stream nat := Cons _ n (from (n + 1)).
```

Ovime je definirana funkcija `from` koja za ulazni argument `n` vraća niz prirodnih brojeva od `n` na dalje.

Razlika induktivnih i koinduktivnih tipova može se izreći epigramom:

„Induktivni tipovi su domene rekurzivnih funkcija, a koinduktivni tipovi su kodomene korekurzivnih

Time se želi reći da se termi induktivnih tipova destruktuiraju u rekurzivnim funkci-

jama, dok se termi koinduktivnih tipova konstruiraju u korekurzivnim funkcijama.

Misao: *Barendregtova kocka, term koji ovisi o termu, term koji ovisi o tipu, tip koji ovisi o termu, tip koji ovisi o tipu*

2.3. Hijerarhija tipova


U usporedbi s tradicionalnim programskim jezicima, Coqov tipski sustav je ekspresivniji jer dopušta tipove koji mogu ovisiti o termima. Takvi tipovi se u Coqu konstruiraju `forall` izrazom. Primjer jednog takvog tipa je „lista duljine n ”, gdje je n neki prirodan broj, a čiji su stanovnici n -torke. Općeniti tip tada glasi „za svaki n , lista duljine n ” — on ovisi o stanovniku drugog tipa (u ovom slučaju, tipa `nat`).

Spomenuli smo da su i tipovi termi te im se može dodijeliti sorta. Postoji li najveći tip, odnosno postoji li tip `Type` čiji su stanovnici svi dobro formirani termi? Prisjetimo se, svaki term ima svoj tip. Kada bi takav `Type` postojao, tada bi vrijedilo `Type : Type`, što može dovesti do paradoksa lašca¹¹. Takav dokazivač teorema bio bi inkonzistentan te bismo njime mogli dokazati kontradikciju, čime dokazivač efektivno gubi svoju svrhu. Umjesto jedne „velike” sorte `Type`, u Coqu postoji niz monotono rastućih sorti `Typen` za sve prirodne brojeve n , takav da vrijedi `Typen : Typem` kad god vrijedi $n < m$. Ilustracija ove **kumulativne hijerarhije tipova** prikazana je na slici 2.1. Dvije najvažnije sorte u Coqu su sorta `Set` i sorta `Prop`.

Naziv `Set` sinonim je za sortu `Type0`, a njeni stanovnici su **mali tipovi**. **Misao:** *NB: Tu je prije pisalo da su stanovnici od `Set` programi, no to nije tako. Stanovnici sorte `Set` su mali tipovi (ne znam kako ih drugačije nazvati), a tek onda su stanovnici malih tipova programi.* Primjerice, tipovi `nat` i `bool` su mali tipovi. Dodatno, funkcije koje primaju i vraćaju male tipove su također mali tipovi. Tada su i produkti, sume, liste i stabla malih tipova također mali tipovi. Intuitivno se može reći da su mali oni tipovi s čijim se stanovnicima može efektivno računati. Stanovnike malih tipova nazivamo **programima**.

Stanovnici sorte `Prop` su **propozicije**. Za razliku od programa, sa propozicijama ne

¹¹Ova rečenica je lažna. U teoriji skupova to je Russellov paradoks, u teoriji tipova to je Girardov paradoks



Slika 2.1. Kumulativna hijerarhija tipova

možemo efektivno računati, već ih moramo dokazivati. Stanovnici propozicija su njihovi **dokazi**.

Za dokazivače teorema, poželjna je mogućnost definicije predikata (propozicija) nad proizvoljnim tipovima. Zbog toga u Coqu prilikom definicije terma sorte Prop možemo raditi kvantifikaciju po proizvoljno velikim tipovima (što uključuje i sortu Prop).

```
1 Inductive isNat : Set -> Prop :=
2 | IsNat : isNat nat.
```

Tako je ovdje isNat predikat nad sortom Set, a u primjeru ispod not je predikat nad sortom Prop.

```
1 Definition not (P : Prop) := P -> False.
```

Ovaj stil kvantifikacije omogućuje nam definiciju proizvoljnih propozicija i propozicijskih veznika. Kažemo da je sorta Prop **nepredikativna**. S druge strane, sorta Set je **predikativna**, to jest *ne dopušta* kvantifikaciju po proizvoljno velikim tipovima. Na

praktičnoj strani, predikativnost ograničava korisnika da prilikom definicije programa smije koristiti samo druge programe.

2.4. Propozicije i tipovi, dokazi i termi

U decimalnom zapisu broja π , barem jedna znamenka pojavljuje se beskonačno mnogo puta. Doista, kada bi se svaka znamenka javljala samo konačno mnogo puta, broj π bio bi racionalan. Međutim, nije jasno *koja* znamenka ima to svojstvo. Da bismo odgovorili na to pitanje, morali bismo prebrojiti *sve* broja π , što nije moguće u konačno mnogo koraka.

Sličnim pitanjima bavili su se logičari dvadesetog stoljeća. *Klasični* logičari bi gornju tvrdnju smjesta prihvatili, dok bi *konstruktivisti* tražili konkretnu znamenku. Između ostalog, ovakva razmatranja rezultirala su fundamentalnim uvidom u povezanost programiranja i dokazivanja. Naime, želimo li dokazati konjunkciju, dovoljno je zasebno dokazati njene konjunkte. S druge strane, želimo li konstruirati par objekata, dovoljno je zapakirati prvi i drugi objekt u konstruktor para. Na sličan način, želimo li dokazati implikaciju, dovoljno je pretpostaviti njen antecedent te pomoću njega dokazati konzekvens. Ako pak želimo konstruirati funkciju, smijemo uzeti jedan argument i pomoću njega konstruirati povratnu vrijednost. Dodatno, nemoguće je dokazati laž, a istina trivijalno vrijedi. S druge strane, ako tip nema konstruktore, nije moguće definirati vrijednost tog tipa. Ako pak tip ima barem jedan konstruktor, tada postoje i njegovi stanovnici. Kroz ove primjere vidimo fenomen **Curry–Howardove korespondencije**, koju možemo sažeti epigramom:

„Propozicije su tipovi, dokazi su programi.”

Time se dokazivanje svodi na programiranje. Pogledi na dvije strane ovog novčića mogu se vidjeti u tablici 2.1.

Za bolju ilustraciju, prikazujemo princip matematičke indukcije u Coqu. Prisjetimo se, za proizvoljni predikat P nad prirodnim brojevima, princip matematičke indukcije glasi:

$$P(0) \wedge (\forall n, P(n) \rightarrow P(n + 1)) \rightarrow \forall n, P(n).$$

Dokazivanje	Programiranje
propozicija	tip
dokaz	program
laž	prazan tip
istina	nastanjen tip
konjunkcija	produktni tip
disjunkcija	zbrojni tip
implikacija	funkcijski tip
univerzalna kvantifikacija	zavisni produkt
egzistencijalna kvantifikacija	zavisni koprodukt

Tablica 2.1. Sličnosti dokazivanja i programiranja

Tvrđnju dokazujemo analizom broja n . Ako je $n = 0$, tvrdnja slijedi iz *baze* indukcije. Ako pak je $n = n' + 1$ za neki n' , tada rekurzivno konstruiramo dokaz za $P(n')$, a konačna tvrdnja slijedi primjenom *koraka* indukcije na rekurzivno konstruirani dokaz.

```

1 Definition nat_ind (P : nat -> Prop)
2   (baza : P 0)
3   (korak : forall n, P n -> P (S n))
4   : forall n, P n :=
5   fix F (n : nat) : P n :=
6     match n with
7     | 0 => baza
8     | S n' => korak n' (F n')
9   end.

```

Za term `nat_ind` kažemo da je dokazni objekt (*proof object*) za tvrdnju matematičke indukcije. Princip matematičke indukcije je samo poseban slučaj **principa indukcije**, kojeg Coq automatski generira za svaki induktivno definiran tip.

2.5. Ograničenja tipskog sustava

Kao što smo već vidjeli, Coqov tipski sustav je ekspresivniji od tipskih sustava tradicionalnih programskih jezika. Međutim, kako bi se sačuvala poželjna svojstva algoritma provjere tipova, ipak se tipski sustav mora ograničiti.

Uvjet pozitivnosti odnosi se na definiciju induktivnih i koinduktivnih tipova. **Mi-sao:** *NB: uvjet pozitivnosti mora vrijediti i u pozitivnim i u negativnim koinduktivnim ti-*

povima. Ovo ograničenje zabranjuje *negativne* pojave tipa kojeg definiramo u argumentima njegovih konstruktora.

```
1 Inductive Lam :=  
2 | LamVar (n : nat)  
3 | LamApp (M N : Lam)  
4 | LamAbs (M : Lam -> Lam).
```

Pokretanje primjera iznad rezultira greškom `Non strictly positive occurrence of "Lam" in "(Lam -> Lam) -> Lam"` — drugim riječima, tip `Lam` se javlja negativno u konstruktoru `LamAbs`, odnosno kao argument funkcije koja je parametar konstruktora. Ovaj uvjet štiti korisnika od inkonzistentnosti, a za točnu definiciju pozitivnosti čitatelja upućujemo na dokumentaciju.¹² Uz uvjet pozitivnosti za induktivne tipove vezan je **uvjet strukturalne rekurzije**. Ovim uvjetom osigurava se totalnost rekurzivno definirane funkcije tako da se argument po kojem je funkcija rekurzivna strukturalno smanjuje u svakom koraku rekurzije.

Uvjet produktivnosti odnosi se na definiciju korekurzivnih funkcija, a dualan je uvjetu strukturalne rekurzije. Ovaj uvjet također štiti korisnika od inkonzistentnosti, a glasi: svaki korekurzivni poziv smije se javljati točno kao izravan argument konstruktora koinduktivnog tipa čiji element definiramo.¹³ Zbog tog uvjeta, iduća definicija nije moguća.

```
1 Set Primitive Projections.  
2 CoInductive NatStream := {  
3   nat_hd : nat;  
4   nat_tl : NatStream;  
5 }.  
6  
7 CoFixpoint foo : NatStream := foo.
```

Greška koju sustav javlja glasi `Unguarded recursive call in "foo"`, što znači da se korekurzivni poziv `foo` *ne* javlja kao izravni argument konstruktora. S druge strane, definicija

```
1 CoFixpoint bar : NatStream := { | nat_hd := 0; nat_tl := bar | }.
```

¹²<https://coq.inria.fr/doc/v8.18/refman/language/core/inductive.html#well-formed-inductive-definitions>

¹³<https://coq.inria.fr/doc/v8.18/refman/language/core/coinductive.html#co-recursive-functions-cofix>

je sasvim legalna. **Misao:** *Tu se baš i ne vidi zašto se bar javlja kao izravan argument konstruktora jer koristimo negativni koinduktivni tip. Ispod haube je to sve zamotano u neki `Build_NatStream`.*

Misao: *Bili smo spomenuli četiri ograničenja na sastanku. Uvjet strukturalne rekurzije i uvjet produktivnosti su na neki način dualni. Treće ograničenje bio je uvjet pozitivnosti. Čovjek očekuje da onda postoji nešto dualno uvjetu pozitivnosti za koinduktivne tipove. Nije li to opet uvjet pozitivnosti?*

Misao: *Ipak neću spomenuti razlike između pozitivnih i negativnih koinduktivnih tipova. Mislim da je to izvan dosega ovog rada.*

Posljednje ograničenje koje spominjemo vezano je uz irelevantnost dokaza (*proof irrelevance*). Naime, mnogi teoremi mogu se dokazati na više načina, ali svaki pojedini dokaz (dakle, postupak kojim smo od pretpostavka došli do konkluzije) *nije bitan*. Matematičarima su bitni samo iskaz teorema i činjenica da se teorem *može dokazati*. Sam postupak dokazivanja smatra se „implementacijskim detaljom”. Upravo zato analiza dokaza ima smisla samo kada se dokazuje, ali ne i kada se programira. U našoj terminologiji to znači da se *pattern matching* nad dokazima smije provoditi samo kod definiranja terma sorte `Prop`. U suprotnom, mogli bismo definirati programe koji ovise o *konkretnom dokazu*, umjesto o iskazanom teoremu. **Ograničenje eliminacije propozicije** nastalo je radi ekstrakcije — svi termini sorte `Prop` se „brišu” prilikom prevođenja iz Coqovog tipskog sustava u tipske sustave niže razine apstrakcije. Kada ovog ograničenja ne bi bilo, ekstrakcija u jednostavnije jezike naprosto ne bi bila moguća.

3. Logika prvog reda s induktivnim definicijama

U ovom poglavlju predstavljamo glavne rezultate diplomskog rada koji uključuju formalizaciju logike prvog reda s induktivnim definicijama FOL_{ID} te dokaznog sustava $LKID$, koje je prvi uveo Brotherston [11]. Definicije, leme i dokazi u ovom poglavlju preuzete su iz Brotherstonove disertacije [12].

Prvo ćemo definirati sintaksu i semantiku logike FOL_{ID} , nakon čega ćemo definirati njene standardne modele. Zatim ćemo prikazati dokazni sustav $LKID$ te konačno dokazati adekvatnost sustava $LKID$ s obzirom na standardnu semantiku, što je ujedno i glavni rezultat ovog diplomskog rada.

Svaka definicija i lema u ovom poglavlju bit će popraćena svojim pandanom u Coqu. Jedan je od ciljeva diplomskog rada prikazati primjene Coqa u matematici, zbog čega leme nećemo dokazivati „na papiru”, već se dokaz svake leme može pronaći na GitHub repozitoriju rada.¹ Zainteresiranom čitatelju predlažemo interaktivni prolazak kroz dokaze lema.

Prije no što krenemo na formalizaciju, valja prokomentirati odnos matematičkog i Coqovog vokabulara. U matematici pojam „skup” može imati dva značenja; prvo se odnosi na skupove kao *domene diskursa*, dok se drugo odnosi na skupove kao *predikate*, odnosno podskupove. Primjerice, skup prirodnih brojeva \mathbb{N} je domena diskursa kada je riječ o svim prirodnim brojevima te zbog toga pišemo $n \in \mathbb{N}$ umjesto $\mathbb{N}(n)$. S druge strane, skup svih parnih brojeva E je podskup skupa \mathbb{N} , a može se interpretirati kao predikat na prirodnim brojevima te pišemo $E(n)$ umjesto $n \in E$. U Coqu se skupovi kao domene diskursa formaliziraju tipovima sorte Set^2 , dok se skupovi kao predikati forma-

¹TODO: REPO LINK

²Ili općenito kao tipovi sorte Type .

liziraju funkcijama iz domene diskursa u sortu `Prop`. Na primjer, tip prirodnih brojeva `nat` je sorte `Set`, a predikat `Nat`. Even je tipa `nat → Prop`.

3.1. Sintaksa

Kao i u svakom izlaganju logike, na početku je potrebno definirati sintaksu.

Definicija 1 (Signatura). *Jezik prvog reda s induktivnim predikatima* (kratko signatura), u oznaci Σ , je skup simbola od kojih razlikujemo *funkcijske*, *obične predikatne* i *induktivne predikatne* simbole. Mjesnost simbola reprezentiramo funkcijom iz odgovarajućeg skupa simbola u skup \mathbb{N} . Funkcijski simboli mjesnosti nula nazivaju se *konstante*, a predikatni simboli mjesnosti nula nazivaju se *propozicije*.

```

1 Structure signature := {
2   FuncS : Set;
3   fun_ar : FuncS -> nat;
4   PredS : Set;
5   pred_ar : PredS -> nat;
6   IndPredS : Set;
7   indpred_ar : IndPredS -> nat
8 }.

```

Primjer 1. Misao: Σ_{PA}

U ostatku poglavlja promatramo jednu proizvoljnu, ali fiksiranu signaturu Σ . Fiksiranje nekog proizvoljnog objekta je česta pojava u matematici, prvenstveno zato što fiksirane argumente ne trebamo spominjati eksplicitno. Coq omogućuje fiksiranje naredbom `Context`, pod uvjetom da se korisnik nalazi u `Section` okolini.³ Većina definicija i lema u ovom radu su napisane upravo unutar `Section` okoline.

Definicija 2 (Term). *Varijabla* je prirodan broj. Skup terma konstruiramo rekurzivno na način:

1. svaka varijabla je term;
2. ako je f funkcijski simbol mjesnosti n te su t_1, \dots, t_n termi⁴, onda je $f(t_1, \dots, t_n)$ također term.

³<https://coq.inria.fr/doc/v8.18/refman/language/core/sections.html>

⁴Primijetimo, broj terma ovisi o mjesnosti funkcijskog simbola. U Coq implementaciji ovog „konstruktor“ možemo vidjeti da je on zavisnog tipa.


```

1 Inductive term : Set :=
2 | var_term : var -> term
3 | TFunc : forall (f : FuncS Σ), vec term (fun_ar f) -> term.

```

Općenitije prezentacije logike prvog reda za skup varijabli uzimaju proizvoljan skup \mathcal{V} , no za formalizaciju je pogodniji skup prirodnih brojeva \mathbb{N} . Kasnije ćemo objasniti zašto.

Princip indukcije za term potrebno je ručno definirati. Naime, induktivni tip term je *ugniježđen* po konstruktoru TFunc što znači da se javlja omotan oko drugog induktivnog tipa⁵ kao argument. Za ugniježdene induktivne tipove, Coq generira *neprikladne* principe indukcije.

```

1 Lemma term_ind
2   : forall P : term Σ -> Prop,
3     (forall v, P (var_term v)) ->
4     (forall f args, (forall st, V.In st args -> P st) ->
5       P (TFunc f args)) ->
6     forall t : term Σ, P t.

```

Definicija 3. Skup svih varijabli koje se javljaju u termu t , u oznaci $TV(t)$, konstruiramo rekurzivno na način:

1. za varijablu v vrijedi $TV(v) = \{v\}$,
2. za n -mjesni funkcijski simbol f i terme t_1, \dots, t_n vrijedi $TV(f(t_1, \dots, t_n)) = \bigcup_{1 \leq i \leq n} TV(t_i)$.

```

1 Inductive TV : term -> var -> Prop :=
2 | TVVar : forall v, TV (var_term v) v
3 | TVFunc : forall f args v st, V.In st args ->
4           TV st v -> TV (TFunc f args) v.

```

Definicija 4 (Formula). Skup formula konstruiramo rekurzivno na način:

1. ako je Q običan ili induktivan predikatni simbol mjesnosti n te su t_1, \dots, t_n termi, onda je $Q(t_1, \dots, t_n)$ *atomarna* formula;
2. ako je φ formula, onda su $\neg\varphi$ i $\forall\varphi$ također formule;
3. ako su φ i ψ formule, onda je $\varphi \rightarrow \psi$ također formula.

⁵Ovdje vec.

```

1 Inductive formula : Set :=
2 | FPred (P : PredS  $\Sigma$ ) : vec (term  $\Sigma$ ) (pred_ar P) -> formula
3 | FIndPred (P : IndPredS  $\Sigma$ ) : vec (term  $\Sigma$ ) (indpred_ar P) -> formula
4 | FNeg : formula -> formula
5 | FImp : formula -> formula -> formula
6 | FAll : formula -> formula.

```

Ostale veznike definiramo kao sintaksne pokrate.

```

1 Definition FAnd ( $\varphi$   $\psi$  : formula) : formula := FNeg (FImp  $\varphi$  (FNeg  $\psi$ )).
2 Definition FOr ( $\varphi$   $\psi$  : formula) : formula := FImp (FNeg  $\varphi$ )  $\psi$ .
3 Definition FExist ( $\varphi$  : formula) : formula := FNeg (FAll (FNeg  $\varphi$ )).

```

Umjesto eksplicitne kvantifikacije po nekoj varijabli v , mi ćemo implicitno kvantificirati po varijabli 0. Ovaj pristup kvantifikaciji⁶, imena „de Bruijnovo indeksiranje”, bitno olakšava rad sa supstitucijama, a uveden je u članku [13]. O samoj implementaciji de Bruijnovog indeksiranja više se može pročitati u knjizi *Types and Programming Languages* [14]. Za potrebe ovog rada koristili smo program *Autosubst2*⁷ [15, 16] za automatsko generiranje tipova terma i formula te pripadajućih funkcija supstitucija i pomoćnih lema.

Misao: Ovaj odlomak preseliti nekamo na početak odjeljka.

Definicija 5. Skup slobodnih varijabli formule φ , u oznaci $FV(\varphi)$, konstruiramo rekurzivno na način:

1. $FV(P(u_1, \dots, u_n)) = \bigcup_{1 \leq i \leq n} TV(u_i)$,
2. $FV(\neg\varphi) = FV(\varphi)$,
3. $FV(\varphi \rightarrow \psi) = FV(\varphi) \cup FV(\psi)$,
4. $FV(\forall\varphi) = \{v \mid 1 + v \in FV(\varphi)\}$.

```

1 Inductive FV : formula -> var -> Prop :=
2 | FV_Pred : forall R args v st,
3   V.In st args -> TV st v -> FV (FPred R args) v
4 | FV_IndPred : forall R args v st,
5   V.In st args -> TV st v -> FV (FIndPred R args) v
6 | FV_Imp_l : forall F G v, FV F v -> FV (FImp F G) v
7 | FV_Imp_r : forall F G v, FV G v -> FV (FImp F G) v
8 | FV_Neg : forall F v, FV F v -> FV (FNeg F) v
9 | FV_All : forall F v, FV F (S v) -> FV (FAll F) v.

```

Definicija 6 (Supstitucija). Supstitucija je svaka funkcija iz skupa \mathbb{N} u skup terma, a

⁶Ili općenitije, vezivanju varijabli.

⁷<https://github.com/uds-psl/autosubst2>

domena joj se može rekurzivno proširiti na skup terma i skup formula.

```

1 Fixpoint subst_term (σ : var -> term) (t : term) : term :=
2   match t with
3   | var_term v => σ v
4   | TFunc f args => TFunc f (V.map (subst_term σ) args)
5   end.

1 Fixpoint subst_formula
2   (σ : var -> term Σ) (φ : formula )
3   : formula :=
4   match φ return formula with
5   | FPred P args => FPred P (V.map (subst_term σ) args)
6   | FIndPred P args => FIndPred P (V.map (subst_term σ) args)
7   | FNeg ψ => FNeg (subst_formula σ ψ)
8   | FImp ψ ξ => FImp (subst_formula σ ψ) (subst_formula σ ξ)
9   | FAll ψ => FAll (subst_formula (up_term_term σ) ψ)
10  end.

```

Ovdje funkcija *up_term_term* brine da supstitucija σ mijenja samo one varijable koje nisu vezane.

Konačno, potrebno je definirati sintaksu za indukciju. U Coqu su definicije induktivnih propozicija proizvoljne do na ograničenje pozitivnosti, no radi jednostavnosti u FOL_{ID} su moguće samo induktivne definicije s atomarnim formulama, a pišemo ih u stilu prirodne dedukcije:

$$\frac{Q_1 \mathbf{u}_1 \dots Q_n \mathbf{u}_n \quad P_1 \mathbf{v}_1 \dots P_m \mathbf{v}_m}{P \mathbf{t}}$$

Ovdje su Q_1, \dots, Q_n obični predikatni simboli, P_1, \dots, P_m i P su induktivni predikatni simboli, a podebljani znakovi predstavljaju vektore terma odgovarajućih duljina.

Definicija 7 (Produkcija). Uređene četvorke

1. listi parova običnih predikatnih simbola i vektora terma odgovarajućih duljina,
 2. listi parova induktivnih predikatnih simbola i vektora terma odgovarajućih duljina,
 3. induktivnog predikatnog simbola P mjesnosti n i
 4. vektora terma duljine n
- nazivamo produkcijama.

```

1 Record production := mkProd {
2   preds : list ({ P : PredS  $\Sigma$  & vec (term  $\Sigma$ ) (pred_ar P) });
3   indpreds : list ({ P : IndPredS  $\Sigma$  & vec (term  $\Sigma$ ) (indpred_ar P) });
4   indcons : IndPredS  $\Sigma$ ;
5   indargs : vec (term  $\Sigma$ ) (indpred_ar indcons);
6 }.

```

Skup induktivnih definicija Φ je skup produkcija. U ostatku rada promatramu neki proizvoljan, ali fiksiran skup induktivnih definicija Φ .

```

1 Definition IndDefSet := production -> Prop.

```

3.2. Semantika

Definicija 8 (Struktura). Struktura prvog reda M je uređena četvorka skupa kojeg nazivamo *nosačem* te interpretacija funkcijskih, običnih predikatnih i induktivnih predikatnih simbola. Funkcijski se simboli interpretiraju kao n -mjesne funkcije, a predikatni simboli kao n -mjesne relacije na nosaču. Koristit ćemo ime strukture kao sinonim za njen nosač, a interpretacije označavamo sa f^M odnosno P^M .

```

1 Structure structure := {
2   domain :> Set;
3   interpF (f : FuncS  $\Sigma$ ) : vec domain (fun_ar f) -> domain;
4   interpP (P : PredS  $\Sigma$ ) : vec domain (pred_ar P) -> Prop;
5   interpIP (P : IndPredS  $\Sigma$ ) : vec domain (indpred_ar P) -> Prop;
6 }.

```

Definicija 9. Neka je M proizvoljna struktura. *Okolina ρ za M je proizvoljna funkcija iz skupa prirodnih brojeva u nosač strukture.*

```

1 Definition env := var -> M.

```

Okolina se može interpretirati kao niz d_0, d_1, d_2, \dots . Tada je *pomaknuta okolina*, u oznaci $d \cdot \rho$, niz d, d_0, d_1, d_2, \dots za neki $d \in M$. Proširenje domene okoline ρ na skup terma zovemo *evaluacijom*.

```

1 Fixpoint eval ( $\rho$  : env) (t : term  $\Sigma$ ) : M :=
2   match t with
3   | var_term x =>  $\rho$  x
4   | TFunc f args => interpF f (V.map (eval  $\rho$ ) args)
5   end.

```

Definicija 10 (Relacija ispunjivosti). Neka je M proizvoljna struktura te ρ okolina za M .

Ispunjivost formule φ u okolini ρ pišemo $\rho \models \varphi$, a definiramo rekurzivno na način:

1. ako je P običan ili induktivan predikatni simbol mjesnosti n te su u_1, \dots, u_n termi, onda vrijedi $\rho \models P(u_1, \dots, u_n)$ ako i samo ako vrijedi $P^M(\rho(u_1), \dots, \rho(u_n))$,
2. vrijedi $\rho \models \neg\varphi$ ako i samo ako ne vrijedi $\rho \models \varphi$,
3. vrijedi $\rho \models \varphi \rightarrow \psi$ ako i samo ako $\rho \models \varphi$ povlači $\rho \models \psi$ i
4. vrijedi $\rho \models \forall\varphi$ ako i samo ako za sve $d \in M$ vrijedi $d \cdot \rho \models \varphi$

```

1 Fixpoint Sat (ρ : env M) (F : formula Σ) : Prop :=
2   match F with
3   | FPred P args => interpP P (V.map (eval ρ) args)
4   | FIndPred P args => interpIP P (V.map (eval ρ) args)
5   | FNeg G => ~ Sat ρ G
6   | FImp F G => Sat ρ F -> Sat ρ G
7   | FAll G => forall d, Sat (d .: ρ) G
8   end.

```

Lema 1. Sintaktička i semantička supstitucija komutiraju pod relacijom ispunjivosti.

```

1 Lemma strong_form_subst_sanity2 :
2   forall (φ : formula Σ) (σ : var -> term Σ)
3     (M : structure Σ) (ρ : env M),
4     ρ ⊨ (subst_formula σ φ) <-> (σ >> eval ρ) ⊨ φ.

```

Misao: Zvuči kul. Je li jasno što je to semantička supstitucija i zašto kažem da komutiraju?

3.3. Standardni modeli

Želimo ograničiti semantička razmatranja na samo one strukture koje „imaju smisla” za induktivne predikate. Prisjetimo se, predikatni simbol P mjesnosti n interpretira se na strukturi M podskupom skupa M^n . Indukciju smatramo dokazivanjem u razinama pa ima smisla promatrati *razine interpretacije* induktivnog predikata, gdje je nulta razina prazan skup, a svaku iduću razinu konstruiramo pomoću produkcija induktivnog skupa definicija i prethodnih razina. Tako je prva razina onaj podskup kojeg možemo dobiti samo jednom „primjenom produkcija”, druga je razina onaj podskup kojeg možemo dobiti pomoću dvije primjene produkcija, i tako dalje. Na taj se način, korak po korak, gradi *smisljena* interpretacija induktivnih predikata. Napominjemo da se zbog mogućih među-

ovisnosti induktivnih predikata razine interpretacije definiraju simultano. Ovaj odjeljak posvećujemo formalizaciji ovih pojmova.

Definicija 11 (Operator skupa definicija). Neka je M proizvoljna struktura te neka je pr proizvoljna produkcija induktivnog skupa definicija Φ u čijoj se konkluziji javlja induktivni predikatni simbol P , primjerice:

$$\frac{Q_1 \mathbf{u}_1 \dots Q_n \mathbf{u}_n \quad P_1 \mathbf{v}_1 \dots P_m \mathbf{v}_m}{P \mathbf{t}}$$

Neka je f funkcija koja svakom induktivnom predikatnom simbolu R pridružuje podskup skupa $M^{|R|}$. Tada je $\varphi_{pr}(f)$ skup svih $|P|$ -torki \mathbf{d} elemenata nosača M za koje postoji okolina ρ za M takva da:

- za sve $i \in \{1, \dots, n\}$ vrijedi $\mathbf{u}_i^\rho \in Q_i^M$,
- za sve $j \in \{1, \dots, m\}$ vrijedi $\mathbf{v}_j^\rho \in f(P_j)$ i
- $\mathbf{d} = \mathbf{t}^\rho$.

```

1 Definition φ_pr
2   (pr : production)
3   (args : forall P : IndPredS Σ, vec D (indpred_ar P) -> Prop)
4   (ds : vec D (indpred_ar (indcons pr)))
5   : Prop :=
6     exists (ρ : env M),
7     (forall Q us, List.In (Q; us) (preds pr) ->
8       interpP Q (V.map (eval ρ) us))
9     /\
10    ( forall P ts, List.In (P; ts) (indpreds pr) ->
11      args P (V.map (eval ρ) ts))
12    /\
13    ds = V.map (eval ρ) (indargs pr).

```

Operator φ_{pr} je formalizacija ideje primjene produkcije. Nadalje, potrebno je definirati operator koji će uzeti u obzir sve produkcije koje se odnose na P . Definiramo $\varphi_P(f)$ kao uniju svih $\varphi_{pr'}(f)$ gdje je pr' produkcija u kojoj se P javlja u konkluziji.

```

1 Definition  $\varphi_P$ 
2   (P : IndPredS  $\Sigma$ )
3   (args : forall P : IndPredS  $\Sigma$ , vec D (indpred_ar P) -> Prop)
4   : vec D (indpred_ar P) -> Prop.
5   refine (fun ds => _).
6   refine (@ex production (fun pr => _)).
7   refine (@ex (P = indcons pr /\  $\Phi$  pr) (fun H => _)).
8   destruct H as [Heq H $\Phi$ ].
9   rewrite Heq in ds.
10  exact ( $\varphi_{pr}$  pr args ds).
11 Defined.

```

Konačno, definiramo operator skupa definicija φ_Φ kao preslikavanje koje induktivnom predikatnom simbolu P pridružuje skup $\varphi_P(f)$.

```

1 Definition  $\varphi_\Phi$ 
2   (args : forall P : IndPredS  $\Sigma$ , vec D (indpred_ar P) -> Prop)
3   : forall P : IndPredS  $\Sigma$ , vec D (indpred_ar P) -> Prop :=
4   fun P =>  $\varphi_P$  P args.

```

Operator φ_Φ omogućuje simultanu primjenu produkcija.

Napomena. Kako je funkcija f bila uvedena na samom početku prethodne definicije, u stvari definicija operatora φ_Φ glasi $\varphi_\Phi(f)(P) = \varphi_P(f)$.

Misao: Ova duga definicija vapi za primjerom. Meni je jasno, ali slutim da bi dragom čitatelju bilo incomprehensible.

Propozicija 1. Operator φ_Φ je monoton.

```

1 Proposition  $\varphi_\Phi\_monotone$  :
2   forall (f g : forall P, vec D (indpred_ar P) -> Prop),
3     (forall P v, f P v -> g P v) ->
4     (forall P v,  $\varphi_\Phi$  f P v ->  $\varphi_\Phi$  g P v).

```

Definicija 12 (Aproksimanti). Neka je M proizvoljna struktura. Definiramo aproksimaciju skupa induktivnih definicija Φ razine α , u oznaci φ_Φ^α , rekursivno na način:

1. $\varphi_\Phi^0(P) = \emptyset$
2. $\varphi_\Phi^{\alpha+1} = \varphi_\Phi(\varphi_\Phi^\alpha)$.

```

1 Fixpoint  $\varphi_{\Phi\_n}$  P ( $\alpha$  : nat) (v : vec M (indpred_ar P)) : Prop :=
2   match  $\alpha$  with
3   | 0 => False
4   | S  $\alpha$  =>  $\text{@}\varphi_{\Phi}$   $\Sigma$  M  $\Phi$  (fun P =>  $\varphi_{\Phi\_n}$  P  $\alpha$ ) P v
5   end.

```

Tada je aproksimant induktivnog predikatnost simbola P razine α upravo $\varphi_{\Phi}^{\alpha}(P)$.

```

1 Definition approximant_of (P : IndPredS  $\Sigma$ )
2   : nat -> vec M (indpred_ar P) -> Prop :=
3    $\varphi_{\Phi\_n}$  P.

```

Misao: Redoslijed argumenata u `approximant_of` je zbunjujuć s obzirom na čitanje simbola: fi fi na alfa od pe. **Misao:** O redoslijedu smo na nekom sastanku raspravljali i zaključili da bi ipak trebao biti ovakav kakav jest.

Napomena. Brotherston je definirao aproksimaciju razine α kao uniju aproksimacija svih nižih razina. Takva je definicija ekvivalentna našoj.

Misao: Malo je neprecizno reći “unija”, ustvari se misli na uniju za svaki pojedini P .

Lema 2. Za sve prirodne brojeve α i induktivne predikatne simbole P vrijedi

$$\varphi_{\Phi}^{\alpha}(P) = \bigcup_{\beta < \alpha} \varphi_{\Phi}^{\beta}(P).$$

```

1 Lemma approximant_characterization : forall a P v,
2    $\varphi_{\Phi\_n}$  P a v <-> exists  $\beta$ ,
3    $\beta < a$  /\  $\text{@}\varphi_{\Phi}$   $\Sigma$  M  $\Phi$  (fun P =>  $\varphi_{\Phi\_n}$  P  $\beta$ ) P v.

```

Definicija 13. Aproksimacija razine ω , u oznaci φ_{Φ}^{ω} , je unija aproksimacija razina manjih od ω .

```

1 Definition  $\varphi_{\Phi\_w}$  P v := exists  $\alpha$ ,  $\varphi_{\Phi\_n}$  P  $\alpha$  v.

```

Misao: Opet, “unija” nije precizno.

Lema 3. Aproksimacija razine ω najmanji je skup sa svojstvom $\varphi_{\Phi}(\varphi_{\Phi}^{\omega}) \subseteq \varphi_{\Phi}^{\omega}$.


```

1 Lemma  $\omega\_prefixed$  : forall P v, @ $\varphi_{\Phi}$   $\Sigma$  M  $\Phi$   $\varphi_{\Phi_{\omega}}$  P v ->  $\varphi_{\Phi_{\omega}}$  P v.
2 Lemma  $\omega\_least$  : forall args,
3     (forall P v, @ $\varphi_{\Phi}$   $\Sigma$  M  $\Phi$  args P v -> args P v) ->
4     forall P v,  $\varphi_{\Phi_{\omega}}$  P v -> args P v.

```

Definicija 14 (Standardni model). Kažemo da je struktura M standardni model za Φ ako interpretira svaki induktivni predikatni simbol njegovim aproksimantom razine ω .

```

1 Definition standard_model
2   ( $\Sigma$  : signature) ( $\Phi$ : @IndDefSet  $\Sigma$ )
3   : structure  $\Sigma$  -> Prop :=
4   fun M =>
5     forall (P : IndPredS  $\Sigma$ ) ts, interpIP P ts <-> @ $\varphi_{\Phi_{\omega}}$   $\Sigma$  M  $\Phi$  P ts.

```

Standardni modeli su upravo one strukture koje imaju smisla za skup induktivnih definicija.

3.4. Sistem sekvenata s induktivnim definicijama

Kako to da je Brotherston definirao signaturu, ali onda nije uključio aksiome u LKID? Primjerice, u Σ_{PA} imamo simbol jednakosti, ali ne možemo dokazati refleksivnost. Valjda su aksiomi uključeni u Γ , u smislu LKID je općenit dokazni sustav, pa bi neka “aksiomatizirana teorija” u stvari bili dokazi *aksiomi* $\models \varphi$. Možda.

LKID. Dopustiva pravila. Primjeri dokaza.

3.5. Adekvatnost

Lokalne adekvatnosti za pravila izvoda. Glavni teorem.

4. Ciklički dokazi

Koinduktivni tip podatka i koinduktivna propozicija. Jedan primjer su Streamovi i predikat `Infinite`. Jednostavniji primjer bi možda bio koinduktivni `nat` i koinduktivni `le`.

Kako bi izgledali ciklički dokazi u LKID? Ono što je tamo “repeat funkcija” je u Coqu `cofix`.

5. Zaključak

Literatura

- [1] Y. Bertot i P. Castéran, *Interactive theorem proving and program development: Coq'Art: the Calculus of Inductive Constructions*. Springer Science & Business Media, 2013.
- [2] B. C. Pierce, A. A. de Amorim, C. Casinghino, M. Gaboardi, M. Greenberg, C. Hrițcu, V. Sjöberg, i B. Yorgey, *Logical Foundations*, ser. Software Foundations, B. C. Pierce, Ur. Electronic textbook, 2023., sv. 1, version 6.5, <http://softwarefoundations.cis.upenn.edu>.
- [3] B. C. Pierce, A. A. de Amorim, C. Casinghino, M. Gaboardi, M. Greenberg, C. Hrițcu, V. Sjöberg, A. Tolmach, i B. Yorgey, *Programming Language Foundations*, ser. Software Foundations, B. C. Pierce, Ur. Electronic textbook, 2024., sv. 2, version 6.5, <http://softwarefoundations.cis.upenn.edu>.
- [4] A. W. Appel, *Verified Functional Algorithms*, ser. Software Foundations, B. C. Pierce, Ur. Electronic textbook, 2023., sv. 3, version 1.5.4, <http://softwarefoundations.cis.upenn.edu>.
- [5] A. Chlipala, *Certified programming with dependent types: a pragmatic introduction to the Coq proof assistant*. MIT Press, 2022.
- [6] The Coq Development Team, “The Coq Reference Manual, Release 8.18.0”, <https://coq.inria.fr/doc/v8.18/refman/>, 2023.
- [7] F. Pfenning i C. Paulin-Mohring, “Inductively Defined Types in the Calculus of Constructions”, u *Proceedings of the 5th International Conference on Mathematical Foundations of Programming Semantics*. Berlin, Heidelberg: Springer-Verlag, 1989., str. 209–228.

- [8] E. Giménez, “Codifying guarded definitions with recursive schemes”, u *Types for Proofs and Programs*, P. Dybjer, B. Nordström, i J. Smith, Ur. Berlin, Heidelberg: Springer Berlin Heidelberg, 1995., str. 39–59.
- [9] M. Sozeau, S. Boulier, Y. Forster, N. Tabareau, i T. Winterhalter, “Coq Coq correct! verification of type checking and erasure for Coq, in Coq”, *Proceedings of the ACM on Programming Languages*, sv. 4, br. POPL, str. 1–28, 2019.
- [10] G. Hutton, *Programming in Haskell*, 2. izd. Cambridge University Press, 2016.
- [11] J. Brotherston, “Cyclic proofs for first-order logic with inductive definitions”, u *Automated Reasoning with Analytic Tableaux and Related Methods*, B. Beckert, Ur. Berlin, Heidelberg: Springer Berlin Heidelberg, 2005., str. 78–92.
- [12] —, “Sequent calculus proof systems for inductive definitions”, doktorska disertacija, School of Informatics, University of Edinburgh, 2006.
- [13] N. G. de Bruijn, “Lambda calculus notation with nameless dummies, a tool for automatic formula manipulation, with application to the Church-Rosser theorem”, *Indagationes Mathematicae (Proceedings)*, sv. 75, br. 5, str. 381–392, 1972.
- [14] B. C. Pierce, *Types and Programming Languages*. MIT Press, 2002.
- [15] K. Stark, “Mechanising Syntax with Binders in Coq”, doktorska disertacija, Saarland University, 2020.
- [16] K. Stark, S. Schäfer, i J. Kaiser, “Autosubst 2: Reasoning with Multi-Sorted de Bruijn Terms and Vector Substitutions”, *8th ACM SIGPLAN International Conference on Certified Programs and Proofs, CPP 2019, Cascais, Portugal, January 14-15, 2019*, 2019.

Sažetak

Primjene Coq alata za dokazivanje u matematici i računarstvu

Miho Hren

Unesite sažetak na hrvatskom.

Ključne riječi: prva ključna riječ; druga ključna riječ; treća ključna riječ

Abstract

Applications of the Coq Proof Assistant in mathematics and computer science

Miho Hren

Enter the abstract in English.

Keywords: the first keyword; the second keyword; the third keyword